

# Using SAS PROC CALIS to fit Level-1 error covariance structures of latent growth models

Cherng G. Ding · Ten-Der Jane

Published online: 15 February 2012  
© Psychonomic Society, Inc. 2012

**Abstract** In the present article, we demonstrate the use of SAS PROC CALIS to fit various types of Level-1 error covariance structures of latent growth models (LGM). Advantages of the SEM approach, on which PROC CALIS is based, include the capabilities of modeling the change over time for latent constructs, measured by multiple indicators; embedding LGM into a larger latent variable model; incorporating measurement models for latent predictors; and better assessing model fit and the flexibility in specifying error covariance structures. The strength of PROC CALIS is always accompanied with technical coding work, which needs to be specifically addressed. We provide a tutorial on the SAS syntax for modeling the growth of a manifest variable and the growth of a latent construct, focusing the documentation on the specification of Level-1 error covariance structures. Illustrations are conducted with the data generated from two given latent growth models. The coding provided is helpful when the growth model has been well determined and the Level-1 error covariance structure is to be identified.

**Keywords** Error covariance structure · Latent growth model · Structural equation modeling

The latent growth model (LGM) plays an important role in repeated measure analysis over a limited number of occasions in large samples (e.g., Meredith & Tisak, 1990; Muthén & Khoo, 1998; Preacher, Wichman, MacCallum, &

Briggs, 2008, p. 12). The model can not only characterize intraindividual (within subjects) change over time, but also examine interindividual (between subjects) difference by means of random growth coefficients, and is a typical application of hierarchical linear modeling (HLM). The within-subjects errors over time and the between-subjects errors are conventionally referred to as “Level-1” and “Level-2” errors, respectively. LGM can also be handled by using structural equation modeling (SEM) (e.g., Bauer, 2003; Bollen & Curran, 2006; Chan, 1998; Curran, 2003; Duncan, Duncan, & Hops, 1996; Mehta & Neal 2005; Meredith & Tisak, 1990; Willet & Sayer, 1994). SEM and HLM stem from different statistical theory, and each has developed its own terminology and standard ways of framing research questions. However, there exists much overlap between the two methodologies under some circumstances. Typically, when a two-level data structure arises from the repeated observations of a variable over time for a set of individuals (so that time is hierarchically nested within each individual), SEM is analytically equivalent to HLM (e.g., Bauer, 2003; Bovaird, 2007; Curran, 2003; MacCallum, Kim, Malarkey, & Kiecolt-Glaser, 1997; Raudenbush, 2001; Rovine & Molenaar, 2000; Willett & Sayer, 1994). The SEM approach provides advantages over the HLM approach in examining model fit; modeling the change over time for latent constructs, with the curve-of-factors model; embedding LGM into a larger latent variable model, with the factor-of-curves model; and incorporating measurement models for latent predictors (e.g., Bauer, 2003; Bollen & Curran, 2006, Chap. 7, 8; Bovaird, 2007; Chan, 1998; Curran, 2003; Duncan, Duncan, & Strycker, 2006, Chap. 4; MacCallum et al., 1997; Raudenbush, 2001; Rovine & Molenaar, 2000). However, the SEM approach suffers from a tedious and error-prone data management task. Many steps are needed to properly structure the data,

C. G. Ding (✉) · T.-D. Jane  
Institute of Business and Management,  
National Chiao Tung University,  
118 Chung-Hsiao West Road, Section 1,  
Taipei, Taiwan  
e-mail: cding@mail.nctu.edu.tw

and the SEM code quickly becomes unwieldy. In contrast, the HLM approach allows for simpler model specification, is computationally more efficient, and can easily be expanded to higher level growth models for manifest variables (Curran, 2003; Wu, West, & Taylor, 2009). A detailed comparison between HLM and SEM can be seen in Bauer (2003) and Curran (2003).

Specialized software for SEM such as EQS (Bentler & Wu, 2005), LISREL (Jöreskog & Sörbom, 2001), *Mplus* (Muthén & Muthén, 2007), *Mx* (Neale, Boker, Xie, & Maes, 2003), and SAS PROC CALIS (SAS Institute Inc., 2010) are readily available. HLM (Raudenbush, Bryk, & Congdon, 2005), MLwiN (Rasbash et al., 2000), and SAS PROC MIXED (SAS Institute Inc., 2010) are typical software for HLM. Because of the isomorphism between SEM and HLM for the same growth model, parameter estimates with SEM and those with HLM should be equivalent. Any minor variations can be attributed to different computational methods used (standard maximum likelihood [ML] estimation or full information maximum likelihood [FIML] estimation for SEM, and restricted maximum-likelihood estimation for HLM). Relevant discussions have been given in Bauer (2003), Bovaird (2007), Curran (2003), and Mehta and Neale (2005).

Level-1 errors could be autocorrelated. Autocorrelations, considered to be nuisance parameters, might result from carryover effects, memory effects, practice effects, or other unmodeled associations, and might not be present when a more complex model or a more appropriate time structure is used (Grimm & Widaman, 2010; Sivo & Fan, 2008). For example, the growth curve ARMA ( $p, q$ ) model has been proposed to absorb error autocorrelations (e.g., Sivo, Fan, & Witt, 2005; Sivo & Fan, 2008). When Level-1 errors are autocorrelated, misspecification of their covariance structures has a substantial impact on the inference for model parameters (Ferron, Dailey, & Yi, 2002; Kwok, West, & Green, 2007; Murphy & Pituch, 2009). However, correct covariance structure is difficult to specify by theory (Kwok et al., 2007, p. 588). Therefore, a specification search becomes needed. Littell, Milliken, Stroup, Wolfinger, & Schabenberger (2006, Chap. 5) illustrated two types of tools with SAS PROC MIXED to help select a covariance structure. First are graphical tools to visualize correlation patterns among residuals. Second are information criteria measuring the relative fit of competing covariance structures. AIC (Akaike, 1974) and BIC (Schwarz, 1978) are commonly used descriptive measures. The model that minimizes AIC or BIC is preferred. Before using these methods, researchers should first rule out covariance structures that are obviously inconsistent with the characteristics of the data. On the other hand, although linear growth curve models are often fitted because of their ease in estimation, theory may suggest that more complex growth models be used, since they can better capture developmen-

tal patterns. Correctly specifying the growth model might lead to a simple covariance structure (Grimm & Widaman, 2010). Moreover, when the growth model is misspecified, statistical inference during the search process can be misleading (Yuan & Bentler, 2004). Therefore, the growth model should be well determined before searching for an “optimal” covariance structure for Level-1 errors.

A variety of processes underlying Level-1 errors may be specified (e.g., Newsom, 2002; Singer & Willett, 2003, Chap. 7; Wolfinger, 1996). SAS PROC MIXED contains more than 30 different types of Level-1 preprogrammed error processes. However, some important processes are unavailable, and any modification of existing processes is not allowed. In contrast, there exists much flexibility in PROC CALIS when specifying error covariance structures. For example, the second-order autoregressive process, not available in PROC MIXED, can be handled with PROC CALIS. The strength of PROC CALIS is always accompanied with technical coding work, which needs to be specifically addressed, and is the focus of this article. In addition to PROC CALIS, any comparable SEM software could be used.

There seems to be no commonly acceptable criteria for assessing model fit according to the indices such as AIC and BIC resulting from PROC MIXED. In contrast, there is some agreement on the cut-off criteria of conventional fit indices based on the likelihood ratio test in SEM, such as RMSEA, CFI, and NNFI (TLI) (e.g., Hu & Bentler, 1999). However, since in SEM-based LGM the factor loadings are usually fixed at time points rather than freely estimated, and the fit of the model to the mean structure should be reflected as well, assessment of model fit by using conventional SEM-based fit indices should be cautious (Mehta & Neale, 2005; Wu et al., 2009). When every individual is observed at the same fixed set of time points (called “balanced”) with no missing values (called “complete”), ML estimation is used; otherwise, FIML estimation is used (Wu et al., 2009). With FIML estimation, the model-implied means and covariances are computed for each individual, and the maximum likelihood chi-square fit function is obtained by summing  $-2 \log$  likelihood across all of the individual data vectors (Bovaird, 2007). For balanced and complete data, FIML simplifies to ML, and, in this case, RMSEA, CFI, and NNFI among the SEM-based fit indices have shown good potential performance in evaluating the fit of LGM (Wu & West, 2010; Wu et al., 2009). For unbalanced designs or missing data, conventional guidelines for adequate fit with these indices may be misleading (Wu et al., 2009).

During the search process, we need instruments for the implementation of fitting various types of error covariance structures. The primary motivation to use PROC CALIS is to take advantage of its flexibility in specifying Level-1 error covariance structures and its capability to deal with growth modeling for both manifest variables and latent constructs. PROC CALIS performs better than PROC


MIXED, but more sophisticated coding work is required. The purpose of the present article is to address this issue by giving a tutorial on the syntax using PROC CALIS to fit many types of Level-1 error covariance structures in LGM for a manifest variable as well as for a latent construct. Illustrations will be conducted with the data generated from two given latent growth models. SAS is a general-purpose and publicly available software. Its ability to do data management and analysis within a single package will make the instruments we provide attractive to many researchers.

**Latent growth models**

In this section, we briefly introduce the LGM with a variety of Level-1 error covariance structures through a typical example depicted in Fig. 1. In the figure,  $y_1 - y_4$  denote the repeated measures of  $y$  on four occasions and  $X$  is a Level-2 predictor.  $\eta_{\alpha_i}$  is the unobserved intercept representing the initial status for individual  $i$ , and  $\eta_{\beta_i}$  the unobserved slope showing the individual’s linear rate of change per unit increase in time.  $\eta_{\alpha_i}$  and  $\eta_{\beta_i}$  are both latent factors. The Level-1 model can be written as

$$y = \Lambda_y^* \eta + \varepsilon, \tag{1}$$

where  $y = [y_1 \ y_2 \ y_3 \ y_4]'$ ,  $\Lambda_y^* = \begin{bmatrix} 1 & 1 & 1 & 1 \\ \lambda_1 & \lambda_2 & \lambda_3 & \lambda_4 \end{bmatrix}'$ ,



**Fig. 1** Linear latent growth model with four repeated measures and a predictor  $X$  (adapted from Bollen and Curran [2006, p. 128] and Preachers et al. [2008, p. 29])

$\eta = [\eta_{\alpha} \ \eta_{\beta}]'$ , and  $\varepsilon = [\varepsilon_1 \ \varepsilon_2 \ \varepsilon_3 \ \varepsilon_4]'$ .  $\lambda_t$  is the measurement time points ( $t = 1, 2, 3, 4$ ) and  $\varepsilon$  denotes Level-1 errors. The solid line with four arrowheads presented in Fig. 1 indicate that  $\varepsilon_t$  are pairwise correlated. The factor loading associated with initial status are all fixed at 1, whereas those associated with the slope are set at the value  $\lambda_t$  to reflect the particular time point  $t$  for individual  $i$ . A common coding of  $\lambda_t$  for different time points is to set  $\lambda_1 = 0$  for baseline and  $\lambda_t = t - 1$  for the follow-ups. For this model, subject  $i$ 's growth trajectory is a straight line,  $\eta_{\alpha_i} + \lambda_t \eta_{\beta_i}$ ,  $\lambda_t = 0, 1, 2, 3$ . (For simplicity, subscript  $i$  is omitted for the rest part of this article.) The loading matrix  $\Lambda_y^*$  containing fixed values has a superscript  $*$  to distinguish from the traditional notation used for the unknown loadings in confirmatory factor analysis (CFA). The model is a restricted CFA model.

The Level-2 model can be written as

$$\eta = \Gamma_0 + \Gamma_x \mathbf{x} + \zeta_{\eta}, \tag{2}$$

where  $\Gamma_0 = [\gamma_{00} \ \gamma_{01}]'$ ,  $\Gamma_x = [\gamma_{10} \ \gamma_{11}]'$ ,  $\mathbf{x} = [X]$ , and  $\zeta_{\eta} = [\zeta_{\eta_{\alpha}} \ \zeta_{\eta_{\beta}}]'$ . Growth factors  $\eta_{\alpha}$  and  $\eta_{\beta}$  (a random intercept and a random slope) are both predicted by a time invariant subject-level covariate  $X$ .  $\gamma_{00}$  and  $\gamma_{10}$  denote, respectively, the intercept and slope of the regression of  $\eta_{\alpha}$  on  $X$ ;  $\gamma_{01}$  and  $\gamma_{11}$  are those of  $\eta_{\beta}$  on  $X$ ; and  $\zeta_{\eta_{\alpha}}$  and  $\zeta_{\eta_{\beta}}$  are Level-2 errors. Two or more time invariant predictors of change may be included. Since it is not our focus, for simplicity, we consider only one predictor here.  $\zeta_{\eta}$  and  $\varepsilon$  are assumed to be uncorrelated. The models can be rewritten in combined form as

$$y = \Lambda_y^* (\Gamma_0 + \Gamma_x \mathbf{x}) + \Lambda_y^* \zeta_{\eta} + \varepsilon, \tag{3}$$

based on which the model-implied mean vector  $\mu$  and the model-implied covariance matrix  $\Sigma$  of the manifest variables  $y_1 - y_4$  and  $X$  can be expressed as functions of the model parameters as follows (Bollen & Curran, 2006, p. 134–135):

$$\mu = \begin{bmatrix} \mu_y \\ \mu_x \end{bmatrix} = \begin{bmatrix} \Lambda_y^* (\Gamma_0 + \Gamma_x \mu_x) \\ \mu_x \end{bmatrix}, \tag{4}$$

$$\Sigma = \begin{bmatrix} \Lambda_y^* (\Gamma_x \Sigma_{xx} \Gamma_x' + \Psi_{\zeta_{\eta}}) \Lambda_y^{*'} + \Theta_{\varepsilon} & \Lambda_y^* \Gamma_x \Sigma_{xx} \\ \Sigma_{xx} \Gamma_x' \Lambda_y^* & \Sigma_{xx} \end{bmatrix}, \tag{5}$$

where  $\Theta_{\varepsilon}$  and  $\Psi_{\zeta_{\eta}}$  denote the variance-covariance matrices of  $\varepsilon$  and  $\zeta_{\eta}$ , respectively, and  $\mu_x$  and  $\Sigma_{xx}$  denote, respectively, the mean vector and the variance-covariance matrix of predictors ( $\mu_x = \mu_X$  and  $\Sigma_{xx} = \sigma_X^2$  for this model, since there is only one predictor).

The Level-1 errors,  $\varepsilon_1$ ,  $\varepsilon_2$ ,  $\varepsilon_3$ , and  $\varepsilon_4$ , are assumed to be normally distributed with zero means. The general error covariance matrix (ECM) is unstructured and is given by

$$\Theta_{\varepsilon} = \begin{bmatrix} \sigma_{\varepsilon_1}^2 & & & \\ \sigma_{\varepsilon_2\varepsilon_1} & \sigma_{\varepsilon_2}^2 & & \\ \sigma_{\varepsilon_3\varepsilon_1} & \sigma_{\varepsilon_3\varepsilon_2} & \sigma_{\varepsilon_3}^2 & \\ \sigma_{\varepsilon_4\varepsilon_1} & \sigma_{\varepsilon_4\varepsilon_2} & \sigma_{\varepsilon_4\varepsilon_3} & \sigma_{\varepsilon_4}^2 \end{bmatrix}. \quad (6)$$

The corresponding option given in SAS PROC MIXED is TYPE = UN. Other types of ECM, with fewer parameters may be desirable. The Level-2 errors  $\zeta_{\eta_{\alpha}}$  and  $\zeta_{\eta_{\beta}}$  are assumed to be normally distributed with zero means. Their covariance matrix is usually specified as unstructured (Murphy & Pituch, 2009):

$$\Psi_{\zeta_{\eta}} = \begin{bmatrix} \sigma_{\zeta_{\eta_{\alpha}}}^2 & \sigma_{\zeta_{\eta_{\alpha}}\zeta_{\eta_{\beta}}} \\ \sigma_{\zeta_{\eta_{\alpha}}\zeta_{\eta_{\beta}}} & \sigma_{\zeta_{\eta_{\beta}}}^2 \end{bmatrix}. \quad (7)$$

#### Types of the Level-1 Error Covariance Structure and SAS Statements

Any type of the Level-1 ECM ( $\Theta_{\varepsilon}$ ) except unstructured can be expressed as a set of linear and/or nonlinear constraints on the parameters involving the covariance structure. SAS PROC MIXED provides a REPEATED statement, in which many types of the Level-1 error covariance structure can be specified through the TYPE = option (e.g., Singer, 1998). However, some important processes, such as higher order autoregressive and moving average processes are not included. Moreover, PROC MIXED cannot handle LGM for constructs.<sup>1</sup> To improve, use PROC CALIS. The STD, COV, and PARAMETERS statements in PROC CALIS can be used together to specify any type of ECM. The STD statement defines variances to estimate for exogenous and error variables. The COV statement defines covariances to estimate for exogenous and error variables. The PARAMETERS statement defines additional parameters that are not specified in the models, and it uses both the original and additional parameters for modeling ECM. In other words, each specific type of ECM is composed of functions of the original and additional parameters. The SAS statements in PROC CALIS for fitting different types of the Level-1 error covariance structures, including AR(1) (the first-order autoregressive), MA(1) (the first-order moving average), ARMA(1,1) (the first-order autoregressive moving average), AR(2)

(the second-order autoregressive), MA(2) (the second-order moving average), ARH(1) (heterogeneous AR(1)), TOEPH (heterogeneous Toeplitz), and UN (unstructured)—with four equally spaced occasions—are summarized in Table 1. AR(1), MA(1), ARMA(1,1), AR(2), and MA(2) are members of the ARMA family. Documentation for LGM with ARMA(1,1), TOEPH, and AR(2) for Level-1 errors is given as follows:

*Example 1: ARMA(1,1)* The ARMA(1,1) process is defined as  $\varepsilon_t = \phi_1\varepsilon_{t-1} + v_t - \theta_1v_{t-1}$ , where  $\phi_1$  denotes the autoregressive parameter,  $\theta_1$  the moving average parameter, and  $v_t$  an i.i.d. disturbance process (Box, Jenkins, & Reinsel, 1994, p. 77). Its interpretation is that the Level-1 error at time  $t$  can be predicted by the Level-1 error at time  $t-1$  and the independent disturbance at time  $t-1$ . The resulting ECM is given by

$$\Theta_{\varepsilon} = \sigma_{\varepsilon}^2 \begin{bmatrix} 1 & & & \\ \rho_1 & 1 & & \\ \rho_2 & \rho_1 & 1 & \\ \rho_3 & \rho_2 & \rho_1 & 1 \end{bmatrix}, \quad (8)$$

where  $\sigma_{\varepsilon}^2$  denotes the common variance of  $\varepsilon_i$ ;  $t = 1, 2, 3, 4$ ; and  $\rho_k$  denotes their autocorrelation coefficient at lag  $k$ , given by  $\rho_1 = \frac{(\phi_1 - \theta_1)(1 - \phi_1\theta_1)}{(1 - 2\phi_1\theta_1 + \theta_1^2)}$ ,  $\rho_k = \phi_1\rho_{k-1}$ ,  $k = 2, 3$ ; with the constraints of  $|\phi_1| < 1$  and  $|\theta_1| < 1$ . Program 1 in Appendix A demonstrates how to use PROC CALIS for modeling LGM with the ARMA(1,1) covariance structure for Level-1 errors and the unstructured covariance for Level-2 errors for four equally spaced time points. The UCOV and AUG options are specified to analyze the mean structures in an uncorrected covariance matrix. The data set to be analyzed is augmented by an intercept variable INTERCEPT that has constant values equal to 1. The LINEQS statement given below is used to specify the Level-1 model (the restricted CFA model) shown in Equation 1 and the Level-2 model shown in Equation 2.

#### LINEQS

```
Y1 = 1 F_Alpha + 0 F_Beta + E1,
Y2 = 1 F_Alpha + 1 F_Beta + E2,
Y3 = 1 F_Alpha + 2 F_Beta + E3,
Y4 = 1 F_Alpha + 3 F_Beta + E4,
F_Alpha = GA00 INTERCEPT + GA01X + D0,
F_Beta = GA10 INTERCEPT + GA11X + D1;
```

where F\_ALPHA and F\_BETA represent latent factors  $\eta_{\alpha_i}$  and  $\eta_{\beta_i}$ . Factor loadings are fixed values (in  $\Lambda_{\eta}^*$ ). Level-1 errors  $\varepsilon_1 - \varepsilon_4$  are named E1–E4, and Level-2 errors  $\zeta_{\eta_{\alpha}}$  and  $\zeta_{\eta_{\beta}}$  are named D0 and D1. GA00, GA01, GA10, and GA11 represent estimates of growth parameters  $\gamma_{00}$ ,  $\gamma_{01}$ ,  $\gamma_{10}$ , and  $\gamma_{11}$ .

<sup>1</sup> Although PROC NLMIXED could be used to fit linear or nonlinear LGM for constructs (e.g., Blozis, 2006), no option is available in the procedure for specifying types of ECM. Relevant coding is laborious.

**Table 1** SAS statements in PROC CALIS for specifying different types of the Level-1 error covariance structure with four occasions

Structure ( $\Theta_\varepsilon$ ) and ECM	Statements in PROC CALIS
<p>AR(1):</p> $\varepsilon_t = \phi_1 \varepsilon_{t-1} + v_t,  \phi_1  < 1;$ $\sigma_\varepsilon^2 \begin{bmatrix} 1 & & & \\ \rho_1 & 1 & & \\ \rho_2 & \rho_1 & 1 & \\ \rho_3 & \rho_2 & \rho_1 & 1 \end{bmatrix},$ $\rho_k = \phi_1^k, k > 0.$	<p><b>STD</b></p> <p>E1=VARE, E2=VARE, E3=VARE, E4=VARE, D0 =VARD0, D1 =VARD1;</p> <p><b>COV</b></p> <p>E1 E2=COV_lag1, E2 E3=COV_lag1, E3 E4=COV_lag1, E1 E3=COV_lag2, E2 E4=COV_lag2, E1 E4=COV_lag3, D0 D1=CD0D1;</p> <p><b>PARAMETERS</b> PHI1; COV_lag1= PHI1*VARE; COV_lag2=(PHI1**2)*VARE; COV_lag3= (PHI1**3) *VARE;</p> <p><b>BOUNDS</b></p> <p>-1. &lt; PHI1 &lt; 1. ;</p>
<p>MA(1):</p> $\varepsilon_t = v_t - \theta_1 v_{t-1},  \theta_1  < 1;$ $\sigma_\varepsilon^2 \begin{bmatrix} 1 & & & \\ \rho_1 & 1 & & \\ 0 & \rho_1 & 1 & \\ 0 & 0 & \rho_1 & 1 \end{bmatrix},$ $\rho_1 = \frac{-\theta_1}{(1+\theta_1^2)}, \rho_k = 0, k > 1.$	<p><b>STD</b></p> <p>E1=VARE, E2=VARE, E3=VARE, E4=VARE, D0 =VARD0, D1 =VARD1;</p> <p><b>COV</b></p> <p>E1 E2=COV_lag1, E2 E3=COV_lag1, E3 E4=COV_lag1, D0 D1=CD0D1;</p> <p><b>PARAMETERS</b> THE1; COV_lag1= (-THE1/(1+ THE1**2))*VARE;</p> <p><b>BOUNDS</b></p> <p>-1. &lt; THE1 &lt; 1. ;</p>
<p>ARMA(1,1):</p> $\varepsilon_t = \phi_1 \varepsilon_{t-1} + v_t - \theta_1 v_{t-1},$ $ \phi_1  < 1,  \theta_1  < 1;$ $\sigma_\varepsilon^2 \begin{bmatrix} 1 & & & \\ \rho_1 & 1 & & \\ \rho_2 & \rho_1 & 1 & \\ \rho_3 & \rho_2 & \rho_1 & 1 \end{bmatrix},$ $\rho_1 = \frac{(\phi_1 - \theta_1)(1 - \phi_1 \theta_1)}{(1 - 2\phi_1 \theta_1 + \theta_1^2)},$ $\rho_k = \phi_1 \rho_{k-1}, k > 1.$	<p><b>STD</b></p> <p>E1=VARE, E2=VARE, E3=VARE, E4=VARE, D0=VARD0, D1=VARD1;</p> <p><b>COV</b></p> <p>E1 E2=COV_lag1, E2 E3=COV_lag1, E3 E4=COV_lag1, E1 E3=COV_lag2, E2 E4=COV_lag2, E1 E4=COV_lag3, D0 D1=CD0D1;</p> <p><b>PARAMETERS</b> PHI1 RHO1; COV_lag1=RHO1*VARE; COV_lag2=PHI1* COV_lag1; COV_lag3=PHI1* COV_lag2;</p> <p><b>BOUNDS</b></p> <p>-1. &lt; PHI1 &lt; 1.;</p>
<p>AR(2):</p> $\varepsilon_t = \phi_1 \varepsilon_{t-1} + \phi_2 \varepsilon_{t-2} + v_t,$ $ \phi_2  < 1, \phi_2 + \phi_1 < 1, \phi_2 - \phi_1 < 1;$ $\sigma_\varepsilon^2 \begin{bmatrix} 1 & & & \\ \rho_1 & 1 & & \\ \rho_2 & \rho_1 & 1 & \\ \rho_3 & \rho_2 & \rho_1 & 1 \end{bmatrix},$ $\rho_0 = 1,$ $\rho_1 = \phi_1 / (1 - \phi_2),$ $\rho_k = \phi_1 \rho_{k-1} + \phi_2 \rho_{k-2}, k > 1.$	<p><b>STD</b></p> <p>E1-E4=4*VARE, D0=VARD0, D1=VARD1;</p> <p><b>COV</b></p> <p>E1 E2=COV_lag1, E2 E3=COV_lag1, E3 E4=COV_lag1, E1 E3=COV_lag2, E2 E4=COV_lag2, E1 E4=COV_lag3, D0 D1=CD0D1;</p> <p><b>PARAMETERS</b> PHI1 PHI2; RHO1= PHI1/(1-PHI2); COV_lag1=RHO1*VARE; COV_lag2=PHI1*COV_lag1+ PHI2 *VARE; COV_lag3=PHI1*COV_lag2+PHI2*COV_lag1;</p> <p><b>LINCON</b></p> <p>PHI2 + PHI1 &lt; 1., PHI2 -PHI1 &lt; 1.;</p> <p><b>BOUNDS</b></p> <p>-1. &lt; PHI2 &lt; 1.;</p>

**Table 1** (continued)

Structure ( $\Theta_\varepsilon$ ) and ECM	Statements in PROC CALIS
<p>MA(2):</p> $\varepsilon_t = v_t - \theta_1 v_{t-1} - \theta_2 v_{t-2},$ $ \theta_2  < 1, \theta_2 + \theta_1 < 1, \theta_2 - \theta_1 < 1;$ $\sigma^2 \begin{bmatrix} 1 & & & \\ \rho_1 & 1 & & \\ \rho_2 & \rho_1 & 1 & \\ 0 & \rho_2 & \rho_1 & 1 \end{bmatrix},$ $\rho_1 = \frac{-\theta_1 + \theta_1 \theta_2}{(1 + \theta_1^2 + \theta_2^2)}, \rho_2 = \frac{-\theta_2}{(1 + \theta_1^2 + \theta_2^2)},$ $\rho_k = 0, k > 2.$	<p><b>STD</b></p> <p>E1=VARE, E2=VARE, E3=VARE, E4=VARE, D0=VARD0, D1=VARD1;</p> <p><b>COV</b></p> <p>E1 E2=COV_lag1, E2 E3=COV_lag1, E3 E4=COV_lag1, E1 E3=COV_lag2, E2 E4=COV_lag2, D0 D1=CD0D1;</p> <p><b>PARAMETERS</b> THE1 THE2;</p> <p>COV_lag1=((-THE1+THE1*THE2)/(1+THE1**2+THE2**2))*VARE; COV_lag2=(-THE2/(1+THE1**2+THE2**2))*VARE;</p> <p><b>LINCON</b></p> <p>THE2+ THE1 &lt; 1., THE2 -THE1 &lt; 1.;</p> <p><b>BOUNDS</b></p> <p>-1. &lt; THE2 &lt; 1. ;</p>
<p>ARH(1) (heterogeneous AR(1)):</p> $\begin{bmatrix} \sigma_{\varepsilon_1}^2 & & & & \\ \sigma_{\varepsilon_2} \sigma_{\varepsilon_1} \rho & \sigma_{\varepsilon_2}^2 & & & \\ \sigma_{\varepsilon_3} \sigma_{\varepsilon_1} \rho^2 & \sigma_{\varepsilon_3} \sigma_{\varepsilon_2} \rho & \sigma_{\varepsilon_3}^2 & & \\ \sigma_{\varepsilon_4} \sigma_{\varepsilon_1} \rho^3 & \sigma_{\varepsilon_4} \sigma_{\varepsilon_2} \rho^2 & \sigma_{\varepsilon_4} \sigma_{\varepsilon_3} \rho & \sigma_{\varepsilon_4}^2 & \end{bmatrix}$	<p><b>STD</b></p> <p>E1=VARE1, E2=VARE2, E3=VARE3, E4=VARE4, D0=VARD0, D1=VARD1;</p> <p><b>COV</b></p> <p>E1 E2=COVE1E2, E1 E3=COVE1E3, E1 E4=COVE1E4, E2 E3=COVE2E3, E2 E4=COVE2E4, E3 E4=COVE3E4, D0 D1=CD0D1;</p> <p><b>PARAMETERS</b> RHO;</p> <p>COVE1E2=SQRT(VARE1)*SQRT(VARE2)*RHO; COVE1E3=SQRT(VARE1)*SQRT(VARE3)*RHO**2; COVE1E4=SQRT(VARE1)*SQRT(VARE4)*RHO**3; COVE2E3=SQRT(VARE2)*SQRT(VARE3)*RHO; COVE2E4=SQRT(VARE2)*SQRT(VARE4)*RHO**2; COVE3E4=SQRT(VARE3)*SQRT(VARE4)*RHO;</p>
<p>TOEPH (heterogeneous Toeplitz):</p> $\begin{bmatrix} \sigma_{\varepsilon_1}^2 & & & & \\ \sigma_{\varepsilon_2} \sigma_{\varepsilon_1} \rho_1 & \sigma_{\varepsilon_2}^2 & & & \\ \sigma_{\varepsilon_3} \sigma_{\varepsilon_1} \rho_2 & \sigma_{\varepsilon_3} \sigma_{\varepsilon_2} \rho_1 & \sigma_{\varepsilon_3}^2 & & \\ \sigma_{\varepsilon_4} \sigma_{\varepsilon_1} \rho_3 & \sigma_{\varepsilon_4} \sigma_{\varepsilon_2} \rho_2 & \sigma_{\varepsilon_4} \sigma_{\varepsilon_3} \rho_1 & \sigma_{\varepsilon_4}^2 & \end{bmatrix}$	<p><b>STD</b></p> <p>E1=VARE1, E2=VARE2, E3=VARE3, E4=VARE4, D0=VARD0, D1=VARD1;</p> <p><b>COV</b></p> <p>E1 E2=COVE1E2, E1 E3=COVE1E3, E1 E4=COVE1E4, E2 E3=COVE2E3, E2 E4=COVE2E4, E3 E4=COVE3E4, D0 D1=CD0D1;</p> <p><b>PARAMETERS</b> RHO1 RHO2 RHO3;</p> <p>COVE1E2=SQRT(VARE1)*SQRT(VARE2)*RHO1; COVE1E3=SQRT(VARE1)*SQRT(VARE3)*RHO2; COVE1E4=SQRT(VARE1)*SQRT(VARE4)*RHO3; COVE2E3=SQRT(VARE2)*SQRT(VARE3)*RHO1; COVE2E4=SQRT(VARE2)*SQRT(VARE4)*RHO2; COVE3E4=SQRT(VARE3)*SQRT(VARE4)*RHO1;</p>
<p>UN:</p> $\begin{bmatrix} \sigma_{\varepsilon_1}^2 & & & & \\ \sigma_{\varepsilon_2} \varepsilon_1 & \sigma_{\varepsilon_2}^2 & & & \\ \sigma_{\varepsilon_3} \varepsilon_1 & \sigma_{\varepsilon_3} \varepsilon_2 & \sigma_{\varepsilon_3}^2 & & \\ \sigma_{\varepsilon_4} \varepsilon_1 & \sigma_{\varepsilon_4} \varepsilon_2 & \sigma_{\varepsilon_4} \varepsilon_3 & \sigma_{\varepsilon_4}^2 & \end{bmatrix}$	<p><b>STD</b></p> <p>E1=VARE1, E2=VARE2, E3=VARE3, E4=VARE4, D0=VARD0, D1=VARD1;</p> <p><b>COV</b></p> <p>E1 E2=COVE1E2, E1 E3=COVE1E3, E1 E4=COVE1E4, E2 E3=COVE2E3, E2 E4=COVE2E4, E3 E4=COVE3E4, D0 D1=CD0D1;</p>

The Level-2 ECM,  $\Psi_{\zeta_\eta} = \begin{bmatrix} \sigma_{\zeta_\eta\alpha}^2 & & \\ \sigma_{\zeta_\eta\alpha} \zeta_\eta\beta & \sigma_{\zeta_\eta\beta}^2 & \\ & & \end{bmatrix}$ , is estimated with type = UN.  $\rho_k$  denotes the autocorrelation coefficient at lag  $k$ . SAS PROC MIXED provides only the options of ARMA(1,1) and AR(1) for the ARMA family.

By Equation 8, Level-1 error variances are equal, their autocovariances at lag 1 are equal, and their autocovariances at lag 2 are equal as well. Level-2 error variances/covariances are unstructured, as shown in Equation 7. Therefore, the STD and COV statements are given as follows:

```
STD
E1 = VARE, E2 = VARE, E3 = VARE, E4 = VARE,
D0 = VARD0, D1 = VARD1;
COV
E1 E2 = COV_lag1, E2 E3 = COV_lag1, E3 E4 = COV_lag1,
E1 E3 = COV_lag2, E2 E4 = COV_lag2,
E1 E4 = COV_lag3,
D0 D1 = COVD0D1;
```

PARAMETERS

```
PHI1 RHO1;
COV_lag1 = RHO1*VARE;
COV_lag2 = PHI1*COV_lag1; /*i.e., COV_lag2 = PHI1*RHO1*VARE; */
COV_lag3 = PHI1*COV_lag2; /*i.e., COV_lag3 = (PHI1**2)*RHO1*VARE; */
```

in which PHI1 and RHO1 represent the estimates of  $\rho_1$  and  $\phi_1$ , defined through their relationships with the autocovariances shown in Equation 8. COV\_lag1 = RHO1\*VARE corresponds to the requirement that the common autocovariance at lag 1 be equal to  $\sigma_\varepsilon^2\rho_1$ . The syntax corresponding to the requirements for the autocovariances at lag 2 ( $=\sigma_\varepsilon^2\phi_1\rho_1$ ) and lag 3 ( $=\sigma_\varepsilon^2\phi_1\rho_2 = \sigma_\varepsilon^2\phi_1^2\rho_1$ ) is given in a similar way.

The constraint of  $|\phi_1| < 1$  is specified by the following BOUNDS statement:

```
BOUNDS
-1. < PHI1 < 1.
```

Example 2: TOEPH The ECM resulting from heterogeneous Toeplitz is given by

$$\Theta_\varepsilon = \begin{bmatrix} \sigma_{\varepsilon_1}^2 & & & & \\ \sigma_{\varepsilon_2}\sigma_{\varepsilon_1}\rho_1 & \sigma_{\varepsilon_2}^2 & & & \\ \sigma_{\varepsilon_3}\sigma_{\varepsilon_1}\rho_2 & \sigma_{\varepsilon_3}\sigma_{\varepsilon_2}\rho_1 & \sigma_{\varepsilon_3}^2 & & \\ \sigma_{\varepsilon_4}\sigma_{\varepsilon_1}\rho_3 & \sigma_{\varepsilon_4}\sigma_{\varepsilon_2}\rho_2 & \sigma_{\varepsilon_4}\sigma_{\varepsilon_3}\rho_1 & \sigma_{\varepsilon_4}^2 & \end{bmatrix}, \quad (9)$$

where  $\sigma_{\varepsilon_t}$  denotes the standard deviation for  $\varepsilon_t$ ,  $t = 1, 2, 3, 4$ ; and  $\rho_k$  the autocorrelation at lag  $k$ ;  $k = 1, 2, 3$ . The Level-1 error variances are unequal, but the autocorrelations at the same lag are equal. The STD and COV statements are given as follows:

in which VARE represents the estimate of the common variance  $\sigma_\varepsilon^2$  of the four Level-1 errors, and VARD0 and VARD1 represent the estimates of the variances,  $\sigma_{\zeta_{\eta_\alpha}}^2$  and  $\sigma_{\zeta_{\eta_\beta}}^2$ , of the two Level-2 errors. COV\_lag1 and COV\_lag2 represent, respectively, the common Level-1 error autocovariance estimates at lag 1 and lag 2. COV\_lag3 is the estimate of the error autocovariance at lag 3. CD0D1 is the estimate of  $\sigma_{\zeta_{\eta_\alpha}\zeta_{\eta_\beta}}$ , the covariance of  $\zeta_{\eta_\alpha}$  and  $\zeta_{\eta_\beta}$ .

Since there exist extra parameters in ECM, they need to be defined, and the work can be achieved by using the PARAMETERS statement given by

```
STD
E1 = VARE1, E2 = VARE2, E3 = VARE3, E4 = VARE4,
D0 = VARD0, D1 = VARD1;
COV
E1 E2 = COVE1E2, E1 E3 = COVE1E3, E1 E4 = COVE1E4,
E2 E3 = COVE2E3, E2 E4 = COVE2E4, E3 E4 = COVE3E4,
D0 D1 = COVD0D1;
```

in which VARE1–VARE4 represent the estimates of the four Level-1 error variances, and VARD0 and VARD1 represent those of the two Level-2 error variances. COVE1E2–COVE3E4 represent the corresponding Level-1 error autocovariance estimates, and COVD0D1 represents the Level-2 error autocovariance estimate. Since the error covariances  $\sigma_{\varepsilon_t\varepsilon_t'}$  of  $\varepsilon_t$  and  $\varepsilon_t'$  are given by  $\sigma_{\varepsilon_t\varepsilon_t'} = \sigma_{\varepsilon_t}\sigma_{\varepsilon_t'}\rho_{\varepsilon_t\varepsilon_t'}$  and the autocorrelations at the same lag are constrained to be equal, the following PARAMETERS statement needs to be added:

```
PARAMETERS
RHO1 RHO2 RHO3;
COVE1E2 = Sqrt(VARE1)*Sqrt(VARE2)*RHO1;
COVE2E3 = Sqrt(VARE2)*Sqrt(VARE3)*RHO1;
COVE3E4 = Sqrt(VARE3)*Sqrt(VARE4)*RHO1;
COVE1E3 = Sqrt(VARE1)*Sqrt(VARE3)*RHO2;
COVE2E4 = Sqrt(VARE2)*Sqrt(VARE4)*RHO2;
COVE1E4 = Sqrt(VARE1)*Sqrt(VARE4)*RHO3;
```

where RHO1, RHO2, and RHO3 are estimates of  $\rho_1$ ,  $\rho_2$ , and  $\rho_3$ . The LINEQS statement used for this example is the same as that given in Example 1.

**Example 3: AR(2)** It is not possible to model AR(2) for Level-1 errors by using PROC MIXED, but the task can be done by using PROC CALIS, with the statements shown in Table 1. The AR(2) process, given by  $\varepsilon_t = \phi_1\varepsilon_{t-1} + \phi_2\varepsilon_{t-2} + v_t$ , where  $\phi_1$  and  $\phi_2$  are autoregressive parameters and  $v_t$  an i.i.d. process (Box et al., 1994, p. 54), leads to the following Level-1 ECM:

$$\sigma_\varepsilon^2 \begin{bmatrix} 1 & & & \\ \rho_1 & 1 & & \\ \rho_2 & \rho_1 & 1 & \\ \rho_3 & \rho_2 & \rho_1 & 1 \end{bmatrix}, \quad (10)$$

where  $\sigma_\varepsilon^2$  denotes the common variance of  $\varepsilon_t$ ,  $t = 1, 2, 3, 4$ , and  $\rho_k$  denotes their autocorrelation at lag  $k$ , given by  $\rho_0 = 1$ ,  $\rho_1 = \phi_1/(1 - \phi_2)$ , and  $\rho_k = \phi_1\rho_{k-1} + \phi_2\rho_{k-2}$ ,  $k = 2, 3$ , with the constraints of  $|\phi_2| < 1$ ,  $\phi_2 + \phi_1 < 1$ , and  $\phi_2 - \phi_1 < 1$ . It follows that the autocovariances at lags 1, 2, and 3, denoted respectively by  $\sigma_1$ ,  $\sigma_2$ , and  $\sigma_3$ , are given by  $\sigma_1 = \rho_1\sigma_\varepsilon^2$ ,  $\sigma_2 = \rho_2\sigma_\varepsilon^2 = \phi_1\rho_1\sigma_\varepsilon^2 + \phi_2\sigma_\varepsilon^2 = \phi_1\sigma_1 + \phi_2\sigma_\varepsilon^2$ , and  $\sigma_3 = \rho_3\sigma_\varepsilon^2 = \phi_1\rho_2\sigma_\varepsilon^2 + \phi_2\rho_1\sigma_\varepsilon^2 = \phi_1\sigma_2 + \phi_2\sigma_1$ . Note that the last two constraints are specified by using the LINCON statement. Relevant SAS statements are given as follows:

```
STD
  E1 - E4 = 4*VARE, /*i.e., E1 = VARE, E2 = VARE,
    E3 = VARE, E4 = VARE*/
  D0 = VARD0, D1 = VARD1;
COV
  E1 E2 = COV_lag1, E2 E3 = COV_lag1, E3 E4 = COV_lag1,
  E1 E3 = COV_lag2, E2 E4 = COV_lag2, E1 E4 = COV_lag3,
  D0 D1 = CD0D1;
PARAMETERS PHI1 PHI2;
  RHO1 = PHI1/(1 - PHI2);
  COV_lag1 = RHO1*VARE;
  COV_lag2 = PHI1*COV_lag1 + PHI2*VARE;
  COV_lag3 = PHI1*COV_lag2 + PHI2*COV_lag1;
LINCON
  PHI2 + PHI1 < 1., PHI2 - PHI1 < 1.;
BOUNDS
  -1. < PHI2 < 1.;
```

In addition to those presented in Table 1, more Level-1 error covariance structures for equally spaced data, including ARMA( $p, q$ ) [autoregressive moving average of order ( $p, q$ )], CS (compound symmetry), TOEP( $q$ ) (Toeplitz with  $q$  bands,  $q = 1, \dots, 4$ , in which the first  $q$  bands of the matrix are to be estimated, setting all higher bands equal to zero), CSH (heterogeneous CS), TOEPH( $q$ ) (heterogeneous Toeplitz with  $q$  bands,  $q = 1, \dots, 4$ ), and UN( $q$ ) (UN with

$q$  bands,  $q = 1, \dots, 4$ ), are summarized in Appendix B. In particular, TOEP(1) indicates i.i.d. Level-1 errors. SAS statements in PROC CALIS for each of them can be obtained in a similar way as shown in Table 1.

The Level-1 error covariance structures displayed in Table 1 and Appendix B are frequently seen in the LGM literature (e.g., Beck & Katz, 1995; Blozis, Harring, & Mels, 2008; Dawson, Gennings, & Carter, 1997; Eyduran & Akbas, 2010; Ferron et al., 2002; Goldstein, Healy, & Rasbash, 1994; Heitjan & Sharma, 1997; Keselman, Algina, Kowalchuk, & Wolfinger, 1998; Kowalchuk & Keselman, 2001; Kwok et al., 2007; Littell, Henry, & Ammerman, 1998; Littell, Rendegast, & Natarajan, 2000; Mansour, Nordheim, & Rutledge, 1985; Murphy & Pituch, 2009; Orhan, Eyduran, & Akbas, 2010; Rovine & Molenaar, 1998; 2000; Singer & Willett, 2003; Chap. 7; Velicer & Fava, 2003; West & Hepworth, 1991; Willett & Sayer, 1994; Wolfinger, 1993, 1996; Wulff & Robinson, 2009). The SAS statements provided can facilitate the implementation of their specification.

## Illustration

An illustration is given based on the data set generated from the linear growth model shown in Fig. 1 with the ARH(1) Level-1 error covariance structure and the UN Level-2 error covariance structure. Population parameters are given in Table 2. The sample size of 300 was used (Muthén & Muthén, 2002). The RANDNORMAL function in SAS PROC IML was used to generate multivariate normal data based on the population model-implied mean vector  $\mu$ , shown in Equation 4, and the population model-implied variance–covariance matrix  $\Sigma$ , shown in Equation 5, of  $y$  and  $x$ . The population mean vector and covariance matrix, as well as sample mean and covariance, are reported in Table 2.

The parameter estimates resulting from fitting ARH(1) with PROC CALIS (the SEM approach) and PROC MIXED (the HLM approach), given in Table 3, are very close and verify each other. Furthermore, the fit results from PROC CALIS (chi-square = 11.076 with  $df = 6$ ,  $p = .086$ ; CFI = .998; NNFI = .996; RMSEA = .05) indicate good model fit.

## Second-order latent growth models

A second-order latent growth model can be a curve-of-factors model or a factor-of-curves model (e.g., Duncan, Duncan, & Strycker, 2006, Chap. 4; Hancock, Kuo, & Lawrence, 2001). The curve-of-factors model is used to



**Table 2** Population parameters of the model in Fig. 1 with the Level-1 error covariance structure of ARH(1) and the sample covariance matrix of  $y_1$ – $y_4$  and  $X$  resulting from a data set of size 300 generated from the model

$$\Lambda_y^* = \begin{bmatrix} 1 & 0 \\ 1 & 1 \\ 1 & 2 \\ 1 & 3 \end{bmatrix}, \Gamma_x = \begin{bmatrix} \gamma_{10} \\ \gamma_{11} \end{bmatrix} = \begin{bmatrix} 4 \\ 6 \end{bmatrix}, \Sigma_{xx} = \sigma_x^2 = 1,$$

$$\mu_x = \mu_X = 0, \Gamma_0 = \begin{bmatrix} \gamma_{00} \\ \gamma_{01} \end{bmatrix} = \begin{bmatrix} 10 \\ 4 \end{bmatrix},$$

$$\Psi_{\xi\eta} = \begin{bmatrix} \sigma_{\xi\eta\alpha}^2 & \\ \sigma_{\xi\eta\alpha\xi\eta\beta}^2 & \sigma_{\xi\eta\beta}^2 \end{bmatrix} = \begin{bmatrix} 15 & \\ 7 & 10 \end{bmatrix},$$

$$\Theta_\varepsilon = \begin{bmatrix} \sigma_{\varepsilon_1}^2 & & & & \\ \sigma_{\varepsilon_2}\sigma_{\varepsilon_1}\rho & \sigma_{\varepsilon_2}^2 & & & \\ \sigma_{\varepsilon_3}\sigma_{\varepsilon_1}\rho^2 & \sigma_{\varepsilon_3}\sigma_{\varepsilon_2}\rho & \sigma_{\varepsilon_3}^2 & & \\ \sigma_{\varepsilon_4}\sigma_{\varepsilon_1}\rho^3 & \sigma_{\varepsilon_4}\sigma_{\varepsilon_2}\rho^2 & \sigma_{\varepsilon_4}\sigma_{\varepsilon_3}\rho & \sigma_{\varepsilon_4}^2 & \end{bmatrix},$$

$$\sigma_{\varepsilon_1}^2 = 36, \sigma_{\varepsilon_2}^2 = 25, \sigma_{\varepsilon_3}^2 = 49, \sigma_{\varepsilon_4}^2 = 64,$$

$$\rho = .7$$

Population model-implied mean vector and covariance matrix (Equations 4 and 5)

	$y_1$	$y_2$	$y_3$	$y_4$	$X$
$y_1$	67.000				
$y_2$	83.000	164.000			
$y_3$	113.580	240.500	388.000		
$y_4$	140.464	312.600	501.200	695.000	
$X$	4.000	10.000	16.000	22.000	1.000
Mean	10.000	14.000	18.000	22.000	.000

Sample mean vector and covariance matrix

	$y_1$	$y_2$	$y_3$	$y_4$	$X$
$y_1$	66.557				
$y_2$	80.505	157.888			
$y_3$	109.910	233.411	385.350		
$y_4$	140.643	307.945	501.530	703.510	
$X$	3.103	8.685	14.137	19.714	.855
Mean	9.834	14.098	17.524	21.167	-.001

investigate the growth trajectory of a construct over time. It incorporates the multiple indicators (items) representing the latent construct observed at different time points into the model. Repeated latent constructs are termed the *first-order factors*, and growth factors (i.e., random intercept and slope) are termed the *second-order factors*. The factor-of-curves model includes higher order common factors for random intercepts and random slopes associated with manifest variables used in LGM. In this model, growth factors are the first-order factors and the underlying common intercept and common slope are the second-order factors, accounting for common developmental patterns. Both the curve-of-factors model and the factor-of-curves model can be well handled by using PROC CALIS.

In this section, the second-order demonstration is given for the curve-of-factors model. The model has several advantages (Blozis, 2006; Preacher et al., 2008; Sayer & Cumsille, 2001). First, the model explicitly recognizes the presence of measurement errors in repeated measures and

captures the growth of repeated constructs adjusted for the presence of these errors. Second, the model allows the separation of variation due to departure from the trajectory (temporal instability) and unique variation due to measurement error (unreliability). Third, the model permits the test of longitudinal factorial invariance.

For example, let latent construct  $F$  be measured by three indicators, observed at four occasions, denoted by  $y_{1t}$ – $y_{3t}$ ,  $t = 1, 2, 3, 4$ . The latent constructs  $F_1$ – $F_4$  at the four occasions are the first-order factors, and the growth factors, denoted by  $\eta_\alpha$  and  $\eta_\beta$ , are the second-order factors. Let  $\xi$ , measured by indicators  $x_1$  –  $x_3$ , be a time-invariant latent predictor for the growth factors. The second-order curve-of-factors, LGM, is pictorially presented in Fig. 2, and can be expressed in matrix form as

$$\begin{aligned} y &= \Lambda_y F + \varepsilon, \\ x &= \Lambda_x \xi + \delta, \\ F &= \Lambda_y \eta + \zeta_F, \\ \eta &= \Gamma_0 + \Gamma_\xi \xi + \zeta_\eta, \end{aligned} \tag{11}$$

**Table 3** Summary of the results by fitting ARH(1) for Level-1 errors and UN for Level-2 errors based on the sample covariance matrix shown in Table 2 by using PROC CALIS and PROC MIXED

Assessment of model fit by							
PROC CALIS						PROC MIXED	
Chi-square	df	$P_r >$ chi-square	CFI	NNFI	RMSEA	AIC	BIC
11.076	6	.086	.998	.996	.050	7825.5	7870.0

Parameter estimates by fitting ARH(1) for $\Theta_\varepsilon$ and UN for $\Psi_{\zeta_\eta}$								
Parameters	Estimates by using PROC CALIS				Estimates by using PROC MIXED			
$\Theta_\varepsilon = \begin{bmatrix} \sigma_{\varepsilon_1}^2 & & & & & & & & \\ \sigma_{\varepsilon_2}\sigma_{\varepsilon_1}\rho & \sigma_{\varepsilon_2}^2 & & & & & & & \\ \sigma_{\varepsilon_3}\sigma_{\varepsilon_1}\rho^2 & \sigma_{\varepsilon_3}\sigma_{\varepsilon_2}\rho & \sigma_{\varepsilon_3}^2 & & & & & & \\ \sigma_{\varepsilon_4}\sigma_{\varepsilon_1}\rho^3 & \sigma_{\varepsilon_4}\sigma_{\varepsilon_2}\rho^2 & \sigma_{\varepsilon_4}\sigma_{\varepsilon_3}\rho & \sigma_{\varepsilon_4}^2 & & & & & \end{bmatrix}$	$\begin{bmatrix} 40.56^{**} & & & & & & & & \\ 25.47^* & 29.37^* & & & & & & & \\ 27.68^* & 31.93^* & 63.84^{**} & & & & & & \\ 24.48 & 28.24 & 56.46^* & 91.79^{**} & & & & & \\ & & \hat{\rho} = .74^{***} & & & & & & \\ & & & & & & & & \hat{\rho} = .74^{***} \end{bmatrix}$	$\begin{bmatrix} 40.43^{**} & & & & & & & & \\ 25.37^a & 29.27^* & & & & & & & \\ 27.59^a & 31.83^a & 63.63^{**} & & & & & & \\ 24.40^a & 28.15^a & 56.27^a & 91.49^{**} & & & & & \\ & & \hat{\rho} = .74^{***} & & & & & & \\ & & & & & & & & \end{bmatrix}$						
$\Psi_{\zeta_\eta} = \begin{bmatrix} \sigma_{\zeta_{\eta\alpha}}^2 & & & \\ \sigma_{\zeta_{\eta\alpha}\zeta_{\eta\beta}} & \sigma_{\zeta_{\eta\beta}}^2 & & \\ \gamma_{00} & \gamma_{01} & & \\ \gamma_{10} & \gamma_{11} & & \end{bmatrix}$	$\begin{bmatrix} 14.53 & & & \\ 9.21^{***} & 8.85^{***} & & \\ & & & \\ & & & \\ & & & \\ 10.18^{***} & 3.86^{***} & & \\ 3.68^{***} & 6.48^{***} & & \end{bmatrix}$	$\begin{bmatrix} 14.48 & & & \\ 9.18^{***} & 8.82^{***} & & \\ & & & \\ & & & \\ & & & \\ 10.17^{***} & 3.86^{***} & & \\ 3.68^{***} & 6.47^{***} & & \end{bmatrix}$						

<sup>a</sup> Test for significance cannot be achieved. \*  $p < .05$ , \*\*  $p < .01$ , \*\*\*  $p < .001$


where  $\mathbf{y} = [y_{11}, y_{21}, y_{31}, y_{12}, y_{22}, y_{32}, y_{13}, y_{23}, y_{33}, y_{14}, y_{24}, y_{34}]'$ ,  $\mathbf{x} = [x_1, x_2, x_3]'$ ,  $\mathbf{F} = [F_1, F_2, F_3, F_4]'$ ,  $\boldsymbol{\eta} = [\eta_\alpha, \eta_\beta]'$ ,  $\boldsymbol{\varepsilon} = [\varepsilon_{11}, \varepsilon_{21}, \varepsilon_{31}, \varepsilon_{12}, \varepsilon_{22}, \varepsilon_{32}, \varepsilon_{13}, \varepsilon_{23}, \varepsilon_{33}, \varepsilon_{14}, \varepsilon_{24}, \varepsilon_{34}]'$ ,  $\boldsymbol{\delta} = [\delta_1, \delta_2, \delta_3]'$ ,  $\boldsymbol{\zeta}_F = [\zeta_{F_1}, \zeta_{F_2}, \zeta_{F_3}, \zeta_{F_4}]'$ , and  $\boldsymbol{\zeta}_\eta = [\zeta_{\eta_\alpha}, \zeta_{\eta_\beta}]'$ .  $\Lambda_{\mathbf{y}}$  and  $\Lambda_{\mathbf{x}}$  in the measurement model denote the loading matrices showing the relations of indicators to their underlying constructs. One of the indicators for each construct is selected as the reference indicator, and its loading is fixed to 1 at each time point for scaling purpose (Blozis, 2006; Chan, 1998; Sayer & Cumsille, 2001).  $\Lambda_{\mathbf{y}}$  denotes the loading matrix (with fixed values) of  $\mathbf{F}$  on  $\boldsymbol{\eta}$ .  $\Gamma_0$  and  $\Gamma_\xi$  denote, respectively, the vector of intercepts and slopes of the regressions of the growth factors  $\boldsymbol{\eta}$  on the latent predictor  $\xi$ .  $\boldsymbol{\varepsilon}$  and  $\boldsymbol{\delta}$  denote, respectively, the measurement errors for  $\mathbf{F}$  and  $\xi$ .  $\boldsymbol{\zeta}_F$  and  $\boldsymbol{\zeta}_\eta$  denote, respectively, the errors reflecting the departure of the repeated latent constructs from the trajectory and the errors associated with the random intercept and slope.  $\boldsymbol{\varepsilon}$  and  $\boldsymbol{\zeta}_F$  are Level-1 errors, and  $\boldsymbol{\delta}$  and  $\boldsymbol{\zeta}_\eta$  are Level-2 errors. The assumptions include the following: (a)  $\boldsymbol{\varepsilon}$ ,  $\boldsymbol{\zeta}_F$ ,  $\boldsymbol{\delta}$ , and  $\boldsymbol{\zeta}_\eta$  are uncorrelated; (b)  $\zeta_{F_1}$ ,  $\zeta_{F_2}$ ,  $\zeta_{F_3}$ , and  $\zeta_{F_4}$  are uncorrelated; (c) the measurement errors associated with different indicators are uncorrelated. However, those associated with the same indicator at different points in time are allowed to covary; (d)  $\zeta_{\eta_\alpha}$  and  $\zeta_{\eta_\beta}$  are correlated

(see, e.g., Blozis, 2006; Bollen & Curran, 2006, p. 249; Preacher et al., 2008, p. 63; Sayer & Cumsille, 2001). The correlated measurement errors are depicted in Fig. 2 by the linkage of three solid lines with four arrowheads, one line for each indicator. On the basis of the aforementioned assumptions, the structures of  $\Psi_{\zeta_F}$  and  $\Theta_\delta$  are both TOEPH(1), the structure of  $\Psi_{\zeta_\eta}$  is UN, and the covariance structure of the correlated measurement errors needs to be identified.

Weak factorial invariance is usually assumed in the second-order LGM to allow meaningful interpretations of growth trajectories. Weak factorial invariance requires the equality of the loadings in the measurement model for the same indicator across time (Blozis, 2006; Bollen & Curran, 2006, p. 255; Chan, 1998; Hancock et al., 2001; Preacher et al., 2008, p. 63; Sayer & Cumsille, 2001).

Program 2 in Appendix A demonstrates using PROC CALIS to fit a second-order linear trajectory model for four equally spaced time points, in which AR(1) is specified for three series:  $\boldsymbol{\varepsilon}_{1t}$ ,  $\boldsymbol{\varepsilon}_{2t}$ , and  $\boldsymbol{\varepsilon}_{3t}$ ,  $t = 1, 2, 3, 4$ ; TOEPH(1) is specified for  $\boldsymbol{\zeta}_F$  and  $\boldsymbol{\delta}$ , and UN is specified for  $\boldsymbol{\zeta}_\eta$ . The LINEQS statement, based on Equation 11, is given below. It is an extended version from that in Program 1 by incorporating the measurement models for  $\mathbf{F}$  and the latent predictor  $\xi$ .

**Fig. 2** A second-order linear latent growth model with one time-invariant latent predictor and four repeated latent constructs, each measured by three indicators (adapted from Chan [1998] and Preachers et al. [2008, p. 63])



LINEQS

$$\begin{aligned}
 Y_{11} &= 1 F_1 + EY_{11}, Y_{21} = LY_{21}F_1 + EY_{21}, Y_{31} = LY_{31}F_1 + EY_{31}, \\
 Y_{12} &= 1 F_2 + EY_{12}, Y_{22} = LY_{22}F_2 + EY_{22}, Y_{32} = LY_{32}F_2 + EY_{32}, \\
 Y_{13} &= 1 F_3 + EY_{13}, Y_{23} = LY_{23}F_3 + EY_{23}, Y_{33} = LY_{33}F_3 + EY_{33}, \\
 Y_{14} &= 1 F_4 + EY_{14}, Y_{24} = LY_{24}F_4 + EY_{24}, Y_{34} = LY_{34}F_4 + EY_{34}, \\
 X_1 &= 1 F_7 + EX_1, X_2 = LX_{27} F_7 + EX_2, X_3 = LX_{37} F_7 + EX_3, \\
 F_1 &= 1 F\_Alpha + 0 F\_Beta + EZF_1, \\
 F_2 &= 1 F\_Alpha + 1 F\_Beta + EZF_2, \\
 F_3 &= 1 F\_Alpha + 2 F\_Beta + EZF_3, \\
 F_4 &= 1 F\_Alpha + 3 F\_Beta + EZF_4, \\
 F\_Alpha &= GA00 INTERCEPT + GA10 F_7 + EZF_5, \\
 F\_Beta &= GA01 INTERCEPT + GA11 F_7 + EZF_6, \\
 F_7 &= F7\_int INTERCEPT + EZF_7;
 \end{aligned}$$

where F1–F4 are the first-order factors at the four occasions, and F\_ALPHA and F\_BETA represent the

second-order latent factors,  $\eta_{\alpha_i}$  and  $\eta_{\beta_i}$ .  $Y_{j,t}$  denotes the observed score on the  $j$ th indicator for  $F$  at occasion  $t$ ,  $j = 1,$



equal, and their error autocovariances at lag 2 are equal as well. Therefore, the STD and COV statements are given as follows:

#### STD

```
EY11 – EY14 = 4*VARE1, EY21 – EY24 = 4*VARE2, EY31 – EY34 = 4*VARE3,
EX1 = VAREX1, EX2 = VAREX2, EX3 = VAREX3,
EZF1 = VARZF1, EZF2 = VARZF2, EZF3 = VARZF3, EZF4 = VARZF4,
EZF5 = VARE_Intercept, EZF6 = VARE_Slope, EZF7 = VARZF7;
```

#### COV

```
/*for the Level-1 measurement errors associated with indicator1 */
EY11 EY12 = COV1_lag1, EY12 EY13 = COV1_lag1, EY13 EY14 = COV1_lag1,
EY11 EY13 = COV1_lag2, EY12 EY14 = COV1_lag2, EY11 EY14 = COV1_lag3,
/*for the Level-1 measurement errors associated with indicator2*/
EY21 EY22 = COV2_lag1, EY22 EY23 = COV2_lag1, EY23 EY24 = COV2_lag1,
EY21 EY23 = COV2_lag2, EY22 EY24 = COV2_lag2, EY21 EY24 = COV2_lag3,
/*for the Level-1 measurement errors associated with indicator3*/
EY31 EY32 = COV3_lag1, EY32 EY33 = COV3_lag1, EY33 EY34 = COV3_lag1,
EY31 EY33 = COV3_lag2, EY32 EY34 = COV3_lag2, EY31 EY34 = COV3_lag3,
/*for the Level-2 errors associated with growth factors*/
EZF5 EZF6 = CZF5ZF6;
```

in which VARE1, VARE2, and VARE3 represent, respectively, the estimates of the common variances  $\sigma_{\varepsilon_1}^2$ ,  $\sigma_{\varepsilon_2}^2$ , and  $\sigma_{\varepsilon_3}^2$ . VAREX1–VAREX3 represent the estimates of variances of  $\delta_1 - \delta_3$ . VARZF1–VARZF4 represent the estimates of variances of  $\zeta_{F_1} - \zeta_{F_4}$ . VARE\_Intercept, VARE\_Slope, and CZF5ZF6 represent, the estimates of variances and covariance of the second-order factor errors  $\zeta_{\eta_\alpha}$  and  $\zeta_{\eta_\beta}$ . VARZF7 represents the estimate of variance of the latent

predictor  $\xi$ . COV1\_lag1, COV1\_lag2, and COV1\_lag3 represent, respectively, the estimates of common autocovariance at lags 1, 2, 3 for  $\varepsilon_{1t}$ . Similarly, COV2\_lag1, COV2\_lag2, and COV2\_lag3 represent those for  $\varepsilon_{2t}$ , and COV3\_lag1, COV3\_lag2, and COV3\_lag3 represent those for  $\varepsilon_{3t}$ .

The following PARAMETERS statement is needed to bring three additional parameters,  $\phi_{1\varepsilon_1}$ ,  $\phi_{1\varepsilon_2}$ , and  $\phi_{1\varepsilon_3}$ , based on Equation 13:

#### PARAMETERS PHI1 PHI2 PHI3;

```
/*for the Level-1 measurement errors associated with indicator1 */
COV1_lag1 = PHI1*VARE1; COV1_lag2 = (PHI1**2)*VARE1;
COV1_lag3 = (PHI1**3)*VARE1;
/* for the Level-1 measurement errors associated with indicator2 */
COV2_lag1 = PHI2*VARE2; COV2_lag2 = (PHI2**2)*VARE2;
COV2_lag3 = (PHI2**3)*VARE2;
/* for the Level-1 measurement errors associated with indicator3 */
COV3_lag1 = PHI3*VARE3; COV3_lag2 = (PHI3**2)*VARE3;
COV3_lag3 = (PHI3**3)*VARE3;
```

in which PHI1, PHI2, and PHI3 represent the estimates of  $\phi_{1\varepsilon_1}$ ,  $\phi_{1\varepsilon_2}$ , and  $\phi_{1\varepsilon_3}$ . COV1\_lag1 = PHI1\*VARE1 corresponds to the requirement that the common autocovariance

at lag 1 for  $\varepsilon_{1t}$  be equal to  $\phi_{1\varepsilon_1}\sigma_{\varepsilon_1}^2$ . COV1\_lag2 = (PHI1\*\*2)\*VARE1 corresponds to the requirement that the common autocovariance at lag 2 be equal to  $\phi_{1\varepsilon_1}^2\sigma_{\varepsilon_1}^2$ . COV1\_lag3 =

(PHI1\*\*3)\*VARE1 corresponds to the requirement that the autocovariance at lag 3 be equal to  $\phi_{1\varepsilon_1}^3 \sigma_{\varepsilon_1}^2$ . The relevant statements for  $\varepsilon_{2t}$  and  $\varepsilon_{3t}$  are given similarly.

The constraints of  $|\phi_{1\varepsilon_1}| < 1$ ,  $|\phi_{1\varepsilon_2}| < 1, |\phi_{1\varepsilon_3}| < 1$  are specified by the following BOUNDS statement:

**BOUNDS**

$$-1. < \text{PHI1} < 1., -1. < \text{PHI2} < 1., -1. < \text{PHI3} < 1.;$$

Under the assumption of weak factorial invariance, the LINCON statement should be added to equalize the

loadings for the same indicator across occasions as follows:

**LINCON**

$$\text{LY21F1} = \text{LY22F2}, \text{LY21F1} = \text{LY23F3}, \text{LY21F3} = \text{LY24F4}, \\ \text{LY31F1} = \text{LY32F2}, \text{LY31F1} = \text{LY33F3}, \text{LY31F3} = \text{LY34F4};$$

**Illustration**

Another illustration is given with another data set of size 300 generated from the second-order LGM in Fig. 2. The

**Table 4** Population parameters of the model in Fig. 2 with the Level-1 error covariance structure of AR(1) and the sample covariance matrix resulting from a data set of size 300 generated from the model

$$\Lambda_y = \begin{bmatrix} \lambda_{y_{11}} & 0 & 0 & 0 \\ \lambda_{y_{21}} & 0 & 0 & 0 \\ \lambda_{y_{31}} & 0 & 0 & 0 \\ 0 & \lambda_{y_{12}} & 0 & 0 \\ 0 & \lambda_{y_{22}} & 0 & 0 \\ 0 & \lambda_{y_{32}} & 0 & 0 \\ 0 & 0 & \lambda_{y_{13}} & 0 \\ 0 & 0 & \lambda_{y_{23}} & 0 \\ 0 & 0 & \lambda_{y_{33}} & 0 \\ 0 & 0 & 0 & \lambda_{y_{14}} \\ 0 & 0 & 0 & \lambda_{y_{24}} \\ 0 & 0 & 0 & \lambda_{y_{34}} \end{bmatrix} = \begin{bmatrix} 1.00 & 0 & 0 & 0 \\ .75 & 0 & 0 & 0 \\ .85 & 0 & 0 & 0 \\ 0 & 1.00 & 0 & 0 \\ 0 & .75 & 0 & 0 \\ 0 & .85 & 0 & 0 \\ 0 & 0 & 1.00 & 0 \\ 0 & 0 & .75 & 0 \\ 0 & 0 & .85 & 0 \\ 0 & 0 & 0 & 1.00 \\ 0 & 0 & 0 & .75 \\ 0 & 0 & 0 & .85 \end{bmatrix},$$

$$\Lambda_y^* = \begin{bmatrix} 1 & T_1 - 1 \\ 1 & T_2 - 1 \\ 1 & T_3 - 1 \\ 1 & T_4 - 1 \end{bmatrix} = \begin{bmatrix} 1 & 0 \\ 1 & 1 \\ 1 & 2 \\ 1 & 3 \end{bmatrix}, \Lambda_x = \begin{bmatrix} \lambda_{x_1} \\ \lambda_{x_2} \\ \lambda_{x_3} \end{bmatrix} = \begin{bmatrix} 1 \\ .75 \\ .70 \end{bmatrix},$$

$$\Gamma_0 = \begin{bmatrix} \gamma_{00} \\ \gamma_{01} \end{bmatrix} = \begin{bmatrix} 12 \\ 1 \end{bmatrix}, \Gamma_\xi = \begin{bmatrix} \gamma_{10} \\ \gamma_{11} \end{bmatrix} = \begin{bmatrix} 6 \\ .5 \end{bmatrix},$$

$$\Phi_\zeta = \sigma_\zeta^2 = 4, \mu_\zeta = 13,$$

$$\Theta_\delta = \text{Diag}[\sigma_{\delta_1}^2 \sigma_{\delta_2}^2 \sigma_{\delta_3}^2] = \text{Diag} [.81 \ .36 \ 1.00],$$

$$\Psi_{\zeta_\eta} = \text{Cov}[\zeta_{\eta\alpha} \ \zeta_{\eta\beta}]'$$

$$= \begin{bmatrix} \sigma_{\zeta_{\eta\alpha}}^2 & \\ \sigma_{\zeta_{\eta\alpha}\zeta_{\eta\beta}} & \sigma_{\zeta_{\eta\beta}}^2 \end{bmatrix} = \begin{bmatrix} .80 & \\ .25 & .60 \end{bmatrix},$$

$$\Psi_{\zeta_F} = \text{Diag}[\sigma_{\zeta_{F_1}}^2 \sigma_{\zeta_{F_2}}^2 \sigma_{\zeta_{F_3}}^2 \sigma_{\zeta_{F_4}}^2],$$

$$= \text{Diag} [.25 \ .36 \ .49 \ .64]$$

$$\Theta_\varepsilon = \text{Cov}[\varepsilon_{11} \ \varepsilon_{12} \ \varepsilon_{13} \ \varepsilon_{14} \ \varepsilon_{21} \ \varepsilon_{22} \ \varepsilon_{23} \ \varepsilon_{24} \ \varepsilon_{31} \ \varepsilon_{32} \ \varepsilon_{33} \ \varepsilon_{34}]'$$

$$= \begin{bmatrix} \sigma_{\varepsilon_1}^2 & & & & & & & & & & & \\ \phi_{1\varepsilon_1} \sigma_{\varepsilon_1}^2 & \sigma_{\varepsilon_1}^2 & & & & & & & & & & \\ \phi_{1\varepsilon_1}^2 \sigma_{\varepsilon_1}^2 & \phi_{1\varepsilon_1} \sigma_{\varepsilon_1}^2 & \sigma_{\varepsilon_1}^2 & & & & & & & & & \\ \phi_{1\varepsilon_1}^3 \sigma_{\varepsilon_1}^2 & \phi_{1\varepsilon_1}^2 \sigma_{\varepsilon_1}^2 & \phi_{1\varepsilon_1} \sigma_{\varepsilon_1}^2 & \sigma_{\varepsilon_1}^2 & & & & & & & & \\ 0 & 0 & 0 & 0 & \sigma_{\varepsilon_2}^2 & & & & & & & \\ 0 & 0 & 0 & 0 & \phi_{1\varepsilon_2} \sigma_{\varepsilon_2}^2 & \sigma_{\varepsilon_2}^2 & & & & & & \\ 0 & 0 & 0 & 0 & \phi_{1\varepsilon_2}^2 \sigma_{\varepsilon_2}^2 & \phi_{1\varepsilon_2} \sigma_{\varepsilon_2}^2 & \sigma_{\varepsilon_2}^2 & & & & & \\ 0 & 0 & 0 & 0 & \phi_{1\varepsilon_2}^3 \sigma_{\varepsilon_2}^2 & \phi_{1\varepsilon_2}^2 \sigma_{\varepsilon_2}^2 & \phi_{1\varepsilon_2} \sigma_{\varepsilon_2}^2 & \sigma_{\varepsilon_2}^2 & & & & \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & \sigma_{\varepsilon_3}^2 & & & \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & \phi_{1\varepsilon_3} \sigma_{\varepsilon_3}^2 & \sigma_{\varepsilon_3}^2 & & \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & \phi_{1\varepsilon_3}^2 \sigma_{\varepsilon_3}^2 & \phi_{1\varepsilon_3} \sigma_{\varepsilon_3}^2 & \sigma_{\varepsilon_3}^2 & \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & \phi_{1\varepsilon_3}^3 \sigma_{\varepsilon_3}^2 & \phi_{1\varepsilon_3}^2 \sigma_{\varepsilon_3}^2 & \phi_{1\varepsilon_3} \sigma_{\varepsilon_3}^2 & \sigma_{\varepsilon_3}^2 \end{bmatrix},$$

$$\phi_{1\varepsilon_1} = .5, \sigma_{\varepsilon_1}^2 = .25, \quad \phi_{1\varepsilon_2} = .7, \sigma_{\varepsilon_2}^2 = .36, \quad \phi_{1\varepsilon_3} = .6, \sigma_{\varepsilon_3}^2 = .40.$$

**Table 4** (continued)

Population model-implied mean vector and covariance matrix															
	$y_{11}$	$y_{21}$	$y_{31}$	$y_{12}$	$y_{22}$	$y_{32}$	$y_{13}$	$y_{23}$	$y_{33}$	$y_{14}$	$y_{24}$	$y_{34}$	$x_1$	$x_2$	$x_3$
$y_{11}$	145.30														
$y_{21}$	108.79	81.95													
$y_{31}$	123.29	92.47	105.20												
$y_{12}$	157.18	117.79	133.49	171.51											
$y_{22}$	117.79	88.59	100.12	128.45	96.69										
$y_{32}$	133.49	100.12	113.71	145.57	109.18	124.14									
$y_{13}$	169.36	126.98	143.91	184.88	138.56	157.04	200.94								
$y_{23}$	126.98	95.41	107.93	138.56	104.17	117.78	150.52	113.25							
$y_{33}$	143.91	107.93	122.46	157.04	117.78	133.72	170.59	127.94	145.40						
$y_{14}$	181.58	136.16	154.32	198.66	148.95	168.81	215.78	161.74	183.30	233.59					
$y_{24}$	136.16	102.25	115.74	148.95	111.89	126.61	161.74	121.56	137.48	175.01	131.61				
$y_{34}$	154.32	115.74	131.26	168.81	126.61	143.63	183.30	137.48	156.05	198.34	148.75	168.99			
$x_1$	24.00	18.00	20.40	26.00	19.50	22.10	28.00	21.00	23.80	30.00	22.50	25.50	4.81		
$x_2$	18.00	13.50	15.30	19.50	14.63	16.58	21.00	15.75	17.85	22.50	16.88	19.13	3.00	2.61	
$x_3$	16.80	12.60	14.28	18.20	13.65	15.47	19.60	14.70	16.66	21.00	15.75	17.85	2.80	2.10	2.96
Mean	90.00	67.50	76.50	97.10	72.83	82.54	104.20	78.15	88.57	111.30	83.48	94.61	13.00	9.75	9.10
Sample mean vector and covariance matrix															
	$y_{11}$	$y_{21}$	$y_{31}$	$y_{12}$	$y_{22}$	$y_{32}$	$y_{13}$	$y_{23}$	$y_{33}$	$y_{14}$	$y_{24}$	$y_{34}$	$x_1$	$x_2$	$x_3$
$y_{11}$	147.01														
$y_{21}$	109.81	82.57													
$y_{31}$	124.47	93.16	105.99												
$y_{12}$	158.66	118.63	134.42	172.82											
$y_{22}$	118.87	89.22	100.79	129.40	97.42										
$y_{32}$	134.12	100.36	113.97	145.95	109.46	123.87									
$y_{13}$	171.53	128.31	145.36	186.96	140.12	157.94	203.88								
$y_{23}$	127.89	95.91	108.45	139.39	104.82	117.87	152.01	113.85							
$y_{33}$	144.92	108.44	123.04	157.91	118.44	133.76	172.06	128.45	145.84						
$y_{14}$	182.10	136.18	154.29	198.88	149.11	168.06	216.81	161.76	183.06	232.49					
$y_{24}$	137.16	102.74	116.25	149.83	112.54	126.66	163.35	122.20	137.99	175.04	132.28				
$y_{34}$	154.50	115.60	131.06	168.81	126.64	142.83	183.98	137.38	155.67	197.21	148.66	167.89			
$x_1$	24.39	18.28	20.68	26.40	19.74	22.29	28.51	21.27	24.07	30.08	22.68	25.53	4.94		
$x_2$	18.66	14.02	15.81	20.15	15.14	17.04	21.80	16.29	18.42	23.11	17.45	19.60	3.10	2.77	
$x_3$	18.01	13.47	15.28	19.36	14.55	16.40	20.93	15.60	17.69	22.20	16.73	18.85	3.07	2.34	3.22
Mean	91.24	68.43	77.64	98.60	73.96	83.90	105.8	79.37	89.96	113.04	84.81	96.12	13.21	9.92	9.30

population parameters with the AR(1) covariance structure for Level-1 error processes  $\varepsilon_{1t}$ ,  $\varepsilon_{2t}$ , and  $\varepsilon_{3t}$  and the sample covariance matrix of  $\mathbf{y}$  and  $\mathbf{x}$  resulting from the simulated data set are presented in Table 4. The RANDNORMAL function in PROC IML was used again to generate multivariate normal data based on the population model-implied mean vector and variance–covariance matrix of  $\mathbf{y}$

and  $\mathbf{x}$  in Fig. 2 (see Appendix C for the derivation). The parameter estimates by fitting AR(1) for  $\varepsilon_{1t}$ ,  $\varepsilon_{2t}$ , and  $\varepsilon_{3t}$  are summarized in Table 5. The resulting parameter estimates are all close to the corresponding population values specified in Table 4, and the model fit is excellent (chi-square = 90.49 with  $df = 109$ ,  $p = .9009$ ; CFI = 1.0; NNFI = 1.0; RMSEA < .0001).





**APPENDIX A**

Sample SAS programs for LGM by using PROC CALIS

## Program 1

A SAS program for the LGM shown in Figure 1 by fitting ARMA(1,1) for Level-1 error covariance structure

```
/* The dataset used for PROC CALIS should be a multi-variable dataset rather than a
multi-record dataset (Singer, 1998) */
```

```
PROC CALIS UCOV AUG;
  LINEQS
    Y1 = 1 F_Alpha + 0 F_Beta + E1,
    Y2 = 1 F_Alpha + 1 F_Beta + E2,
    Y3 = 1 F_Alpha + 2 F_Beta + E3,
    Y4 = 1 F_Alpha + 3 F_Beta + E4,
    F_Alpha = GA00 INTERCEPT + GA01 X + D0,
    F_Beta = GA10 INTERCEPT + GA11 X + D1;
  STD
    E1=VARE, E2=VARE, E3=VARE, E4=VARE, X=VARX,
    D0=VARD0, D1=VARD1;
  COV
    E1 E2=COV_lag1, E2 E3=COV_lag1, E3 E4=COV_lag1,
    E1 E3=COV_lag2, E2 E4=COV_lag2, E1 E4=COV_lag3,
    D0 D1=COVD0D1;
  PARAMETERS PHI1 RHO1;
    COV_lag1=RHO1*VARE;
    COV_lag2=PHI1* COV_lag1;
    COV_lag3=PHI1* COV_lag2;
  BOUNDS
    -1. < PHI1 < 1.;
  VAR Y1 Y2 Y3 Y4 X;
  TITLE 'Linear Growth Modeling with Four Occasions by Specifying';
  TITLE2 'ARMA(1,1) for Level-1 Error Covariance Structure';
RUN;
```

## Program 2

A SAS program for the second-order LGM shown in Figure 2 by fitting AR(1) for the Level-1 error covariance structure associated with each indicator

```

PROC CALIS UCOV AUG;
  LINEQS
    Y11 = 1 F1 + EY11,   Y21 = LY21F1 F1 + EY21,   Y31 = LY31F1 F1 + EY31,
    Y12 = 1 F2 + EY12,   Y22 = LY22F2 F2 + EY22,   Y32 = LY32F2 F2 + EY32,
    Y13 = 1 F3 + EY13,   Y23 = LY23F3 F3 + EY23,   Y33 = LY33F3 F3 + EY33,
    Y14 = 1 F4 + EY14,   Y24 = LY24F4 F4 + EY24,   Y34 = LY34F4 F4 + EY34,
    X1 = 1 F7 + EX1,     X2 = LX2F7 F7 + EX2,     X3 = LX3F7 F7 + EX3,
    F1 = 1 F_Alpha + 0 F_Beta + EZF1,
    F2 = 1 F_Alpha + 1 F_Beta + EZF2,
    F3 = 1 F_Alpha + 2 F_Beta + EZF3,
    F4 = 1 F_Alpha + 3 F_Beta + EZF4,
    F_Alpha = GA00 INTERCEPT + GA10 F7 + EZF5,
    F_Beta = GA01 INTERCEPT + GA11 F7 + EZF6,
    F7 = F7_int INTERCEPT + EZF7;
  STD
    EY11–EY14=4*VARE1, EY21–EY24=4*VARE2, EY31–EY34=4*VARE3,
    EX1=VAREX1, EX2=VAREX2, EX3=VAREX3,
    EZF1=VARZF1, EZF2=VARZF2, EZF3=VARZF3, EZF4=VARZF4,
    EZF5=VARE_Intercept, EZF6=VARE_Slope, EZF7=VARZF7;
  COV
    /* for the Level-1 errors associated with indicator 1 */
    EY11 EY12=COV1_lag1, EY12 EY13=COV1_lag1, EY13 EY14=COV1_lag1,
    EY11 EY13=COV1_lag2, EY12 EY14=COV1_lag2, EY11 EY14=COV1_lag3,
    /* for the Level-1 errors associated with indicator 2 */
    EY21 EY22=COV2_lag1, EY22 EY23=COV2_lag1, EY23 EY24=COV2_lag1,
    EY21 EY23=COV2_lag2, EY22 EY24=COV2_lag2, EY21 EY24=COV2_lag3,
    /* for the Level-1 errors associated with indicator 3 */
    EY31 EY32=COV3_lag1, EY32 EY33=COV3_lag1, EY33 EY34=COV3_lag1,
    EY31 EY33=COV3_lag2, EY32 EY34=COV3_lag2, EY31 EY34=COV3_lag3,
    /* for the Level-2 errors associated with growth factors */
    EZF5 EZF6=CZF5ZF6;
  PARAMETERS PHI1 PHI2 PHI3;
    /* for the Level-1 errors associated with indicator 1 */
    COV1_lag1=PHI1*VARE1; COV1_lag2=(PHI1**2)*VARE1;
    COV1_lag3=(PHI1**3)*VARE1;

```

```

/* for the Level-1 errors associated with indicator 2 */
COV2_lag1=PHI2*VARE2; COV2_lag2=(PHI2**2)*VARE2;
COV2_lag3=(PHI2**3)*VARE2;
/* for the Level-1 errors associated with indicator 3 */
COV3_lag1=PHI3*VARE3; COV3_lag2=(PHI3**2)*VARE3;
COV3_lag3=(PHI3**3)*VARE3;
BOUNDS
-1.< PHI1<1., -1.< PHI2<1., -1.< PHI3<1. ;
LINCON /* Weak factorial invariance across time is assumed */
LY21F1=LY22F2, LY21F1=LY23F3, LY21F3=LY24F4,
LY31F1=LY32F2, LY31F1=LY33F3, LY31F3=LY34F4;
TITLE 'Second-Order Linear Growth Modeling for a Construct Measured by';
TITLE2 'Three Indicators at Four Occasions by Fitting AR(1) for the';
TITLE3 'Level-1 Error Covariance Structure Associated with Each Indicator';
VAR Y11 Y21 Y31 Y12 Y22 Y32 Y13 Y23 Y33 Y14 Y24 Y34 X1 X2 X3;
RUN;

```

## APPENDIX B

More types of Level-1 error covariance structures with four equally spaced occasions

Structure ( $\Theta_\varepsilon$ )	ECM
<p>ARMA(<math>p, q</math>):</p> $\varepsilon_t = \phi_1 \varepsilon_{t-1} + \dots + \phi_p \varepsilon_{t-p} + v_t - \theta_1 v_{t-1} - \dots - \theta_q v_{t-q}$	$\sigma_\varepsilon^2 \begin{bmatrix} 1 & & & \\ \rho_1 & 1 & & \\ \rho_2 & \rho_1 & 1 & \\ \rho_3 & \rho_2 & \rho_1 & 1 \end{bmatrix}$
<p>ARMA(1,2):</p> $\varepsilon_t = \phi_1 \varepsilon_{t-1} + v_t - \theta_1 v_{t-1} - \theta_2 v_{t-2},$ $ \phi_1  < 1,  \theta_2  < 1, \theta_1 + \theta_2 < 1, \theta_2 - \theta_1 < 1;$ $\rho_0 = 1, \rho_1 = \frac{(-\theta_1 + \theta_1 \theta_2) + [\phi_1(1 + \theta_2^2) + \theta_1(\theta_1^2 - \theta_2) - \theta_1^2 \theta_1(1 - \theta_2) - \phi_1^2 \theta_2]}{(1 + \theta_1^2 + \theta_2^2) - [2\phi_1(\theta_1 - \theta_1 \theta_2 + \phi_1 \theta_2)]},$ $\rho_2 = \frac{-\theta_2 - [\phi_1 \theta_1(1 - \theta_2) + \theta_1^2(1 + \theta_1^2 + \theta_2^2) - \phi_1^2 \theta_1(1 - \theta_2) - \phi_1^2 \theta_2]}{(1 + \theta_1^2 + \theta_2^2) - [2\phi_1(\theta_1 - \theta_1 \theta_2 + \phi_1 \theta_2)]},$ $\rho_k = \phi_1 \rho_{k-1}, k > 2.$	
<p>ARMA(2,1):</p> $\varepsilon_t = \phi_1 \varepsilon_{t-1} + \phi_2 \varepsilon_{t-2} + v_t - \theta_1 v_{t-1},$ $ \phi_2  < 1, \phi_2 + \phi_1 < 1, \phi_2 - \phi_1 < 1,  \theta_1  < 1;$ $\rho_0 = 1, \rho_1 = [(\phi_1 - \theta_1)(1 - \phi_1 \theta_1) + \phi_2^2 \theta_1] / [(1 - \phi_2)(1 + \theta_1^2) - 2\phi_1 \theta_1],$ $\rho_k = (\phi_1 \rho_{k-1} + \phi_2 \rho_{k-2}), k > 1.$	
<p>CS:</p> $\sigma_\varepsilon^2 [\rho 1(t \neq t') + 1(t = t')],$ $\rho_k = \rho = \sigma_1 / \sigma_\varepsilon^2, k > 0.$	$\begin{bmatrix} \sigma_\varepsilon^2 & & & \\ \sigma_1 & \sigma_\varepsilon^2 & & \\ \sigma_1 & \sigma_1 & \sigma_\varepsilon^2 & \\ \sigma_1 & \sigma_1 & \sigma_1 & \sigma_\varepsilon^2 \end{bmatrix}$

(continued)

Structure ( $\Theta_\varepsilon$ )

ECM

TOEP( $q$ ) (Toeplitz with  $q$  bands,  $q = 1, \dots, 4$ ):

$$\sigma_{|t-t'|} 1(|t-t'| < q), \sigma_0 = \sigma_\varepsilon^2, \sigma_k = \rho_k \sigma_\varepsilon^2, k > 0.$$

$$\begin{array}{cc} \text{TOEP(1)} & \text{TOEP(2)} \\ \begin{bmatrix} \sigma_\varepsilon^2 & & & \\ 0 & \sigma_\varepsilon^2 & & \\ 0 & 0 & \sigma_\varepsilon^2 & \\ 0 & 0 & 0 & \sigma_\varepsilon^2 \end{bmatrix} & , \begin{bmatrix} \sigma_\varepsilon^2 & & & \\ \sigma_1 & \sigma_\varepsilon^2 & & \\ 0 & \sigma_1 & \sigma_\varepsilon^2 & \\ 0 & 0 & \sigma_1 & \sigma_\varepsilon^2 \end{bmatrix} \end{array}$$

$$\begin{array}{cc} \text{TOEP(3)} & \text{TOEP(4)} \\ \begin{bmatrix} \sigma_\varepsilon^2 & & & \\ \sigma_1 & \sigma_\varepsilon^2 & & \\ \sigma_2 & \sigma_1 & \sigma_\varepsilon^2 & \\ 0 & \sigma_2 & \sigma_1 & \sigma_\varepsilon^2 \end{bmatrix} & , \begin{bmatrix} \sigma_\varepsilon^2 & & & \\ \sigma_1 & \sigma_\varepsilon^2 & & \\ \sigma_2 & \sigma_1 & \sigma_\varepsilon^2 & \\ \sigma_3 & \sigma_2 & \sigma_1 & \sigma_\varepsilon^2 \end{bmatrix} \end{array}$$

CSH (heterogeneous CS):

$$\sigma_{\varepsilon_t} \sigma_{\varepsilon_{t'}} [\rho 1(t \neq t') + 1(t = t')],$$

$$\rho_k = \rho, k > 0.$$

$$\begin{bmatrix} \sigma_{\varepsilon_1}^2 & & & \\ \sigma_{\varepsilon_2} \sigma_{\varepsilon_1} \rho & \sigma_{\varepsilon_2}^2 & & \\ \sigma_{\varepsilon_3} \sigma_{\varepsilon_1} \rho & \sigma_{\varepsilon_3} \sigma_{\varepsilon_2} \rho & \sigma_{\varepsilon_3}^2 & \\ \sigma_{\varepsilon_4} \sigma_{\varepsilon_1} \rho & \sigma_{\varepsilon_4} \sigma_{\varepsilon_2} \rho & \sigma_{\varepsilon_4} \sigma_{\varepsilon_3} \rho & \sigma_{\varepsilon_4}^2 \end{bmatrix}$$

TOEPH( $q$ ) (heterogeneous Toeplitz with  $q$  bands,  $q = 1, \dots, 4$ ):

$$\sigma_{\varepsilon_t} \sigma_{\varepsilon_{t'}} \rho_{|t-t'|} 1(|t-t'| < q)$$

$$\begin{array}{cc} \text{TOEPH(1)} & \text{TOEPH(2)} \\ \begin{bmatrix} \sigma_{\varepsilon_1}^2 & & & \\ 0 & \sigma_{\varepsilon_2}^2 & & \\ 0 & 0 & \sigma_{\varepsilon_3}^2 & \\ 0 & 0 & 0 & \sigma_{\varepsilon_4}^2 \end{bmatrix} & , \begin{bmatrix} \sigma_{\varepsilon_1}^2 & & & \\ \sigma_{\varepsilon_2} \sigma_{\varepsilon_1} \rho_1 & \sigma_{\varepsilon_2}^2 & & \\ 0 & \sigma_{\varepsilon_3} \sigma_{\varepsilon_2} \rho_1 & \sigma_{\varepsilon_3}^2 & \\ 0 & 0 & \sigma_{\varepsilon_4} \sigma_{\varepsilon_3} \rho_1 & \sigma_{\varepsilon_4}^2 \end{bmatrix} \end{array}$$

$$\begin{array}{c} \text{TOEPH(3)} \\ \begin{bmatrix} \sigma_{\varepsilon_1}^2 & & & \\ \sigma_{\varepsilon_2} \sigma_{\varepsilon_1} \rho_1 & \sigma_{\varepsilon_2}^2 & & \\ \sigma_{\varepsilon_3} \sigma_{\varepsilon_1} \rho_2 & \sigma_{\varepsilon_3} \sigma_{\varepsilon_2} \rho_1 & \sigma_{\varepsilon_3}^2 & \\ 0 & \sigma_{\varepsilon_4} \sigma_{\varepsilon_2} \rho_2 & \sigma_{\varepsilon_4} \sigma_{\varepsilon_3} \rho_1 & \sigma_{\varepsilon_4}^2 \end{bmatrix} \end{array}$$

$$\begin{array}{c} \text{TOEPH(4)} \\ \begin{bmatrix} \sigma_{\varepsilon_1}^2 & & & \\ \sigma_{\varepsilon_2} \sigma_{\varepsilon_1} \rho_1 & \sigma_{\varepsilon_2}^2 & & \\ \sigma_{\varepsilon_3} \sigma_{\varepsilon_1} \rho_2 & \sigma_{\varepsilon_3} \sigma_{\varepsilon_2} \rho_1 & \sigma_{\varepsilon_3}^2 & \\ \sigma_{\varepsilon_4} \sigma_{\varepsilon_1} \rho_3 & \sigma_{\varepsilon_4} \sigma_{\varepsilon_2} \rho_2 & \sigma_{\varepsilon_4} \sigma_{\varepsilon_3} \rho_1 & \sigma_{\varepsilon_4}^2 \end{bmatrix} \end{array}$$

UN( $q$ ) (UN with  $q$  bands,  $q = 1, \dots, 4$ ):

$$\sigma_{\varepsilon_t \varepsilon_{t'}} 1(|t-t'| < q)$$

$$\begin{array}{cc} \text{UN(1)} & \text{UN(2)} \\ \begin{bmatrix} \sigma_{\varepsilon_1}^2 & & & \\ 0 & \sigma_{\varepsilon_2}^2 & & \\ 0 & 0 & \sigma_{\varepsilon_3}^2 & \\ 0 & 0 & 0 & \sigma_{\varepsilon_4}^2 \end{bmatrix} & , \begin{bmatrix} \sigma_{\varepsilon_1}^2 & & & \\ \sigma_{\varepsilon_2 \varepsilon_1} & \sigma_{\varepsilon_2}^2 & & \\ 0 & \sigma_{\varepsilon_3 \varepsilon_2} & \sigma_{\varepsilon_3}^2 & \\ 0 & 0 & \sigma_{\varepsilon_4 \varepsilon_3} & \sigma_{\varepsilon_4}^2 \end{bmatrix} \end{array}$$

$$\begin{array}{cc} \text{UN(3)} & \text{UN(4)} \\ \begin{bmatrix} \sigma_{\varepsilon_1}^2 & & & \\ \sigma_{\varepsilon_2 \varepsilon_1} & \sigma_{\varepsilon_2}^2 & & \\ \sigma_{\varepsilon_3 \varepsilon_1} & \sigma_{\varepsilon_3 \varepsilon_2} & \sigma_{\varepsilon_3}^2 & \\ 0 & \sigma_{\varepsilon_4 \varepsilon_2} & \sigma_{\varepsilon_4 \varepsilon_3} & \sigma_{\varepsilon_4}^2 \end{bmatrix} & , \begin{bmatrix} \sigma_{\varepsilon_1}^2 & & & \\ \sigma_{\varepsilon_2 \varepsilon_1} & \sigma_{\varepsilon_2}^2 & & \\ \sigma_{\varepsilon_3 \varepsilon_1} & \sigma_{\varepsilon_3 \varepsilon_2} & \sigma_{\varepsilon_3}^2 & \\ \sigma_{\varepsilon_4 \varepsilon_1} & \sigma_{\varepsilon_4 \varepsilon_2} & \sigma_{\varepsilon_4 \varepsilon_3} & \sigma_{\varepsilon_4}^2 \end{bmatrix} \end{array}$$

$1(A)$  equals 1 when  $A$  is true. For example,  $1(|t-t'| < q) = 1$  when  $|t-t'| < q$  and 0 otherwise,  $q \geq 1$ .  $\rho_k$  denotes the autocorrelation coefficient at lag  $k$ .  $\rho_0 = 1$ . TOEP(4) = TOEP; TOEPH(4) = TOEPH; UN(4) = UN; TOEPH(1) = UN(1)

## APPENDIX C

Derivation of the population model-implied mean vector  $\mu$  and variance–covariance matrix  $\Sigma$  of  $y$  and  $x$  for the second-order LGM (Fig. 2)

Based on Equation 11, we have

$$F = \Lambda_y^*(\Gamma_0 + \Gamma_\xi \xi + \zeta_\eta) + \zeta_F = \Lambda_y^* \Gamma_0 + \Lambda_y^* \Gamma_\xi \xi + \Lambda_y^* \zeta_\eta + \zeta_F,$$

$$\begin{aligned} y &= \Lambda_y F + \varepsilon \\ &= \Lambda_y (\Lambda_y^* \Gamma_0 + \Lambda_y^* \Gamma_\xi \xi + \Lambda_y^* \zeta_\eta + \zeta_F) + \varepsilon \\ &= \Lambda_y \Lambda_y^* \Gamma_0 + \Lambda_y \Lambda_y^* \Gamma_\xi \xi + \Lambda_y \Lambda_y^* \zeta_\eta + \Lambda_y \zeta_F + \varepsilon, \end{aligned}$$

$$x = \Lambda_x \xi + \delta.$$

Therefore

$$\mu_y = E(y) = \Lambda_y \Lambda_y^* \Gamma_0 + \Lambda_y \Lambda_y^* \Gamma_\xi \mu_\xi$$

$$\mu_x = E(x) = \Lambda_x \mu_\xi,$$

that is,

$$\mu = \begin{bmatrix} \mu_y \\ \mu_x \end{bmatrix} = \begin{bmatrix} \Lambda_y \Lambda_y^* \Gamma_0 + \Lambda_y \Lambda_y^* \Gamma_\xi \mu_\xi \\ \Lambda_x \mu_\xi \end{bmatrix}$$

and

$$\begin{aligned} \Sigma_{yy} &= Cov(y) = \Lambda_y \Lambda_y^* \Gamma_\xi \Phi_\xi \Gamma_\xi' \Lambda_y^* \Lambda_y' + \Lambda_y \Lambda_y^* \Psi_{\zeta_\eta} \Lambda_y^* \Lambda_y' + \Lambda_y \Psi_{\zeta_F} \Lambda_y' + \Theta_\varepsilon \\ &= \Lambda_y (\Lambda_y^* \Gamma_\xi \Phi_\xi \Gamma_\xi' \Lambda_y^* + \Lambda_y^* \Psi_{\zeta_\eta} \Lambda_y^* + \Psi_{\zeta_F}) \Lambda_y' + \Theta_\varepsilon, \end{aligned}$$

$$\Sigma_{xx} = Cov(x) = \Lambda_x \Phi_\xi \Lambda_x' + \Theta_\delta$$

$$\Sigma_{yx} = Cov(y, x) = \Lambda_y \Lambda_y^* \Gamma_\xi \Phi_\xi \Lambda_x',$$

$$\Sigma_{xy} = \Lambda_x \Phi_\xi' \Gamma_\xi' \Lambda_y^* \Lambda_y'$$

that is

$$\Sigma = \begin{bmatrix} \Sigma_{yy} & \Sigma_{yx} \\ \Sigma_{xy} & \Sigma_{xx} \end{bmatrix} = \begin{bmatrix} \Lambda_y (\Lambda_y^* \Gamma_\xi \Phi_\xi \Gamma_\xi' \Lambda_y^* + \Lambda_y^* \Psi_{\zeta_\eta} \Lambda_y^* + \Psi_{\zeta_F}) \Lambda_y' + \Theta_\varepsilon & \Lambda_y \Lambda_y^* \Gamma_\xi \Phi_\xi \Lambda_x' \\ \Lambda_x \Phi_\xi' \Gamma_\xi' \Lambda_y^* \Lambda_y' & \Lambda_x \Phi_\xi \Lambda_x' + \Theta_\delta \end{bmatrix}.$$

## References

- Akaike, H. (1974). A new look at the statistical model identification. *IEEE Transaction on Automatic Control*, *19*, 716–723.
- Bauer, D. J. (2003). Estimating multilevel linear models as structural models. *Journal of Educational and Behavioral Statistics*, *28*, 135–167.
- Beck, N., & Katz, J. N. (1995). What to do (and not to do) with time-series cross-section data. *American Political Science Review*, *89*, 634–647.
- Bentler, P. M., & Wu, E. J. C. (2005). *EQS 6.1: Structural equation modeling software for windows*. Encino, CA: Multivariate Software, Inc
- Blozis, S. A. (2006). A second-order structured latent curve model for longitudinal data. In K. van Montfort, H. Oud, & A. Satorra (Eds.), *Longitudinal models in the behavioral and related sciences* (pp. 189–214). Mahwah, NJ: Erlbaum.
- Blozis, S. A., Harring, J. R., & Mels, G. (2008). Using LISREL to fit nonlinear latent curve models. *Structural Equation Modeling*, *15*, 346–369.
- Bollen, K. A., & Curran, P. J. (2006). *Latent curve models: A structural equation perspective*. Hoboken, NJ: Wiley.
- Bovaird, J. A. (2007). Multilevel structural equation models for contextual factors. In T. D. Little, J. A. Bovaird, & N. A. Card (Eds.), *Modeling contextual effects in longitudinal studies* (pp. 149–182). Mahwah, NJ: Erlbaum.
- Box, G. E. P., Jenkins, G. M., & Reinsel, G. C. (1994). *Time series analysis: Forecasting and control* (3rd ed.). Englewood Cliffs, NJ: Prentice-Hall.
- Chan, D. (1998). The conceptualization and analysis of change over time: An integrative approach incorporating longitudinal mean and covariance structures analysis (LMACS) and multiple indicator latent growth modeling (MLGM). *Organizational Research Methods*, *1*, 421–483.
- Curran, P. J. (2003). Have multilevel models been structural equation models all along? *Multivariate Behavioral Research*, *38*, 529–569.
- Dawson, K. S., Gennings, C., & Carter, W. H. (1997). Two graphical techniques useful in detecting correlation structure in repeated measures data. *The American Statistician*, *51*, 275–283.

- Duncan, S. C., Duncan, T. E., & Hops, H. (1996). Analysis of longitudinal data within accelerated longitudinal design. *Psychological Methods, 1*, 236–248.
- Duncan, T. E., Duncan, S. C., & Strycker, L. A. (2006). *An introduction to latent variable growth curve modeling: Concepts, issues, and applications* (2nd ed.). Mahwah, NJ: Erlbaum.
- Eyduran, E., & Akbas, Y. (2010). Comparison of different covariance structure used for experimental design with repeated measurement. *Journal of Animal & Plant Sciences, 20*, 44–51.
- Ferron, J., Dailey, R. F., & Yi, Q. (2002). Effects of misspecifying the first-level error structure in two-level models of change. *Multivariate Behavioral Research, 37*, 379–403.
- Goldstein, H., Healy, M. J. R., & Rasbash, J. (1994). Multilevel time series models with applications to repeated measures data. *Statistics in Medicine, 13*, 1643–1655.
- Grimm, K. J., & Widaman, K. F. (2010). Residual structures in latent growth curve modeling. *Structural Equation Modeling, 17*, 424–442.
- Hancock, G. R., Kuo, W. L., & Lawrence, F. R. (2001). An illustration of second-order latent growth models. *Structural Equation Modeling, 8*, 470–489.
- Heitjan, D. F., & Sharma, D. (1997). Modeling repeated series longitudinal data. *Statistics in Medicine, 16*, 347–355.
- Hu, L., & Bentler, P. M. (1999). Cutoff criteria for fit indexes in covariance structure analysis: Conventional criteria versus new alternatives. *Structural Equation Modeling, 6*, 1–55.
- Jöreskog, K., & Sörbom, D. (2001). *LISREL 8: User's reference guide*. Chicago, IL: Scientific Software International.
- Keselman, H. J., Algina, J., Kowalchuk, R. K., & Wolfinger, R. D. (1998). A comparison of two approaches for selecting covariance structures in the analysis of repeated measurements. *Communications in Statistics: Simulation and Computation, 27*, 591–604.
- Kowalchuk, R. K., & Keselman, H. J. (2001). Mixed-model pairwise multiple comparisons of repeated measures means. *Psychological Methods, 6*, 282–296.
- Kwok, O. M., West, S. G., & Green, S. B. (2007). The impact of misspecifying the within-subject covariance structure in multi-wave longitudinal multilevel models: A Monte Carlo study. *Multivariate Behavioral Research, 42*, 557–592.
- Littell, R. C., Henry, P. R., & Ammerman, C. B. (1998). Statistical analysis of repeated measures data using SAS procedures. *American Society of Animal Science, 76*, 1216–1231.
- Littell, R. C., Milliken, G. A., Stroup, W. W., Wolfinger, R. D., & Schabenberger, O. (2006). *SAS for mixed models* (2nd ed.). Cary, NC: SAS Institute Inc
- Littell, R. C., Rensdergast, J., & Natarajan, R. (2000). Modeling covariance structure in the analysis of repeated measures data. *Statistics in Medicine, 19*, 1793–1819.
- MacCallum, R. C., Kim, C., Malarkey, W. B., & Kiecolt-Glaser, J. K. (1997). Studying multivariate change using multilevel models and latent curve models. *Multivariate Behavioral Research, 32*, 215–253.
- Mansour, H., Nordheim, E. V., & Rutledge, J. J. (1985). Maximum likelihood estimation of variance components in repeated measures designs assuming autoregressive errors. *Biometrics, 41*, 287–294.
- Mehta, P. D., & Neale, M. C. (2005). People are variables too: Multilevel structural equations modeling. *Psychological Methods, 10*, 259–284.
- Meredith, W., & Tisak, J. (1990). Latent curve analysis. *Psychometrika, 55*, 107–122.
- Murphy, D. L., & Pituch, K. A. (2009). The performance of multilevel growth curve models under an autoregressive moving average process. *The Journal of Experimental Education, 77*, 255–282.
- Muthén, B. O., & Khoo, S. T. (1998). Longitudinal studies of achievement growth using latent variable modeling. *Learning and Individual Differences, 10*, 73–101.
- Muthén, L. K., & Muthén, B. O. (2002). How to use a Monte Carlo study to decide on sample size and determine power. *Structural Equation Modeling, 9*, 599–620.
- Muthén, L. K., & Muthén, B. O. (2007). *Mplus user's guide* (5th ed.). Los Angeles, CA: Author
- Neale, M. C., Boker, S. M., Xie, G., & Maes, H. H. (2003). *Mx: Statistical modeling* (6th ed.). Richmond, VA: Virginia Commonwealth University.
- Newsom, J. T. (2002). A multilevel structural equation model for dyadic data. *Structural Equation Modeling, 9*, 431–447.
- Orhan, H., Eyduran, E., & Akbas, Y. (2010). Defining the best covariance structure for sequential variation on live weights of anatolian merinos male lambs. *Journal of Animal & Plant Sciences, 20*, 158–163.
- Preacher, K. J., Wichman, A. L., MacCallum, R. C., & Briggs, N. E. (2008). *Latent growth curve modeling*. Thousand Oaks, CA: Sage.
- Rasbash, J., Browne, W. J., Goldstein, H., Yang, M., Plewis, I., Healy, M., Woodhouse, G., Draper, D., Langford, I., & Lewis, T. (2000). *A user's guide to MLwiN* (Version 2.1). London, England: Institute of Education, University of London.
- Raudenbush, S. W. (2001). Comparing personal trajectories and drawing causal inferences from longitudinal data. *Annual Review of Psychology, 52*, 501–525.
- Raudenbush, S. W., Bryk, A. S., & Congdon, R. (2005). *HLM for Windows* (Version 6.03) [Computer Software]. Lincolnwood, IL: Scientific Software International
- Rovine, M. J., & Molenaar, P. C. M. (1998). A LISREL model for the analysis of repeated measure with a patterned covariance matrix. *Structural Equation Modeling, 5*, 318–343.
- Rovine, M. J., & Molenaar, P. C. M. (2000). A structural modeling approach to a multilevel random coefficients model. *Multivariate Behavioral Research, 35*, 51–88.
- SAS Institute Inc. (2010). *SAS/STAT user's guide* (SAS 9.2). Cary, NC: Author
- Sayer, A. G., & Cumsille, P. E. (2001). Second-order latent growth models. In L. Collins & A. G. Sayer (Eds.), *New methods for the analysis of change* (pp. 179–200). Washington DC: American Psychological Association.
- Schwarz, G. (1978). Estimating the dimension of a model. *The Annals of Statistics, 6*, 461–464.
- Singer, J. D. (1998). Using SAS PROC MIXED to fit multilevel models, hierarchical models, and individual growth models. *Journal of Educational and Behavioral Statistics, 24*, 323–355.
- Singer, J. D., & Willett, J. B. (2003). *Applied longitudinal data analysis: Modeling change and event occurrence*. New York, NY: Oxford University Press.
- Sivo, S., & Fan, X. (2008). The latent curve ARMA (p, q) panel model: Longitudinal data analysis in educational research and evaluation. *Educational Research and Evaluation, 14*, 363–376.
- Sivo, S., Fan, X., & Witta, L. (2005). The biasing effects of unmodeled ARMA time series processes on latent growth curve model estimates. *Structural Equation Modeling, 12*, 215–231.
- Velicer, W. F., & Fava, J. L. (2003). Time series analysis. In J. A. Schinka & W. F. Velicer (Eds.), *Handbook of psychology (Research methods in psychology)*, Vol. 2, pp. 581–606). Hoboken, NJ: Wiley.
- West, S. G., & Hepworth, J. T. (1991). Data analytic strategies for temporal data and daily events. *Journal of Personality, 59*, 609–662.

- Willett, J. B., & Sayer, A. G. (1994). Using covariance structure analysis to detect correlates and predictors of individual change over time. *Psychological Bulletin*, *116*, 363–381.
- Wolfinger, R. (1993). Covariance structure selection in general mixed models. *Communications in Statistics—Simulation and Computation*, *22*, 1079–1106.
- Wolfinger, R. (1996). Covariance structures for repeated measures. *Journal of Agricultural, Biological, and Environmental Statistics*, *1*, 205–230.
- Wu, W., & West, S. G. (2010). Sensitivity of fit indices to misspecification in growth curve models. *Multivariate Behavioral Research*, *45*, 420–452.
- Wu, W., West, S. G., & Taylor, A. B. (2009). Evaluating model fit for growth curve models: Integration of fit indices from SEM and MLM frameworks. *Psychological Methods*, *14*, 183–201.
- Wulff, S. S., & Robinson, T. J. (2009). Assessing the uncertainty of regression estimates in a response surface model for repeated measures. *Quality Technology & Quantitative Management*, *6*, 309–324.
- Yuan, K., & Bentler, P. M. (2004). One chi-square difference and  $z$  tests in mean and covariance structure analysis when the base model is misspecified. *Educational and Psychological Measurement*, *64*, 737–757.